MAPLE for Stochastic Differential Equations^{�y}

Sasha Cyganowski

School of Computing and Mathematics Deakin University, Geelong 3217, Australia e-mail: sash@deakin.edu.au

Peter E. Kloeden and Thomas Pohl

Fachbereich Mathematik, Johann Wolfgang Goethe Universität D–60054 Frankfurt am Main, Germany e-mail: kloeden@math.uni-frankfurt.de

�The main part of this report was written while P.E. Kloeden was the Deputy Leader of the Stochastic Algorithms and Nonparametric Statistics Research Group at the Weierstrass Institute.

y1991 Mathematics Subject Classification Primary: 60H10, 60H30, 68Q40 Secondary: 65H05, 93E15, 93E30 i

Contents

1 Introduction 2 1.1 MAPLE . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 2 1.2 Stochastic Differential Equations . . . . . . . . . . . . . . . . . . . . . . . . 3

2 Explicitly Solvable Scalar SDE 4 2.1 Linearsde routine . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 4 2.2 Reducible routine . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 6 2.3 Explicit routine . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 8

3 Ito Stochastic Calculus 9 3.1 LJ Operator routine . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 9 3.2 L0 Operator routine . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 10 3.3 LFP Operator: Fokker–Planck Equation . . . . . . . . . . . . . . . . . . . . 12 3.4 Application of the partial differential operators . . . . . . . . . . . . . . . . 14 3.5 The Ito Formula . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 15 3.6 Coloured Noise . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 16

4 Stratonovich Stochastic Caculus 18 4.1 Ito–Stratonovich correction . . . . . . . . . . . . . . . . . . . . . . . . . . . 18 4.2 Ito–Stratonovich correction: both directions . . . . . . . . . . . . . . . . . . 19 4.3 Stratonovich L0 operator . . . . . . . . . . . . . . . . . . . . . . . . . . . . 20 4.4 Stratonovich transformation formula . . . . . . . . . . . . . . . . . . . . . . 22

5 Linear Vector SDEs 23 5.1 Linearization . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 23 5.2 Second moment equation . . . . . . . . . . . . . . . . . . . . . . . . . . . . 24 5.3 Spherical coordinates . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 27

6 Strong Numerical Schemes 29 6.1 Euler scheme . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 30 6.2 Milstein scheme . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 31 6.3 Order 1.5 strong stochastic Taylor scheme . . . . . . . . . . . . . . . . . . . 33 6.4 2nd order stochastic Taylor scheme . . . . . . . . . . . . . . . . . . . . . . . 35

7 Commutative Noise 37 7.1 Commutative noise of 1st kind . . . . . . . . . . . . . . . . . . . . . . . . . 38 7.2 Commutative noise of 2nd kind . . . . . . . . . . . . . . . . . . . . . . . . . 39

8 Weak Numerical Schemes 41 8.1 Weak Euler scheme . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 41 8.2 Second order weak Taylor scheme . . . . . . . . . . . . . . . . . . . . . . . . 43 8.3 Order 3 weak Taylor scheme . . . . . . . . . . . . . . . . . . . . . . . . . . . 45

9 APPENDIX 47 9.1 Subprocedures for momenteqn . . . . . . . . . . . . . . . . . . . . . . . . . . 47 9.2 The inverse procedure pvector2pmatrix . . . . . . . . . . . . . . . . . . . . . 49

References 50

1 Introduction

1.1 MAPLE

This report provides an introduction and description of the MAPLE software package with(stochastic) which consists of MAPLE routines for stochastic calculus and stochastic differential equations. These are known to work for MapleV version 5 (for Windows) and MapleV version 5 (for Unix). Its CALLING SEQUENCE is

<function>(args) stochastic[<function>](args)

It was originally developed by Sasha Cyganowski of Deakin University, Australia, initially as a Mathematics Honours project under the supervision of Professor P.E. Kloeden and then further extended and developed, partly with the assistance of Dr. Thomas Pohl. Comments via e-mail are welcome and can be sent to Sasha Cyganowski at sash@deakin.edu.au or to Peter Kloeden at kloeden@math.unifrankfurt.de. A copy of the software can be obtained by email from either of these addresses.

The stochastic package contains routines useful for finding explicit solutions of Stochastic Differential Equations(SDEs), and routines useful for constructing numerical schemes up to strong order 2.0 and weak order 3.0. Other features include a routine which converts SDEs with white noise into coloured noise form, and routines which check whether an SDE has commutative noise of the first and/or second kind. Some other useful routines are also available, in particular for the operators procedures L0 and LJ are included so that users can easily construct numerical schemes other than those already available. For more information on a particular function type the command <function>.

The general schemes, solutions and conditions used in the coding of the stochastic package can be found in

Kloeden, P.E, Platen, E.: Numerical Solution of Stochastic Differential Equations, (Springer-Verlag, 1992, Second Edition, 1994).

To use a stochastic function, either define that function alone using the command with(stochastic, <function>), or define all stochastic functions using the command with(stochastic). Alternatively, invoke the function using the long form stochastic[<function>]. This long form notation is necessary whenever there is a conflict between a package function name and another function used in the same session.

The functions available are:

linearsde MLJ explicit L0 Euler
Milstein Taylor1hlf SLO correct Taylor2
wkeuler wktay2 reducible wktay3 colour
comm1 comm2 LJ momenteqn sphere
pmatrix2pvector pvector2pmatrix ap pa bpb

As an example, to find the explicit solution of an SDE with drift coefficient 1/2a^2x and diffusion coefficient ax use

>with(stochastic,explicit); explicit}}(1/2*a^2*x,a*x);

Note the stochastic numerical routines require the drift and diffusion coefficients of an SDE to be entered in the variables x[N] and t, where x[N] are the state variables of the N-dimensional SDE and t denotes time. The routines linearsde, reducible, and explicit require the drift and diffusion coefficients to be entered in the variables x and t.

1.2 Stochastic Differential Equations

We consider a the N-dimensional Ito SDE with an M-dimensional Wiener process dX_t = a(t, X_t) dt + M∑ j=1 b^j(t, X_t) dW^j_t (1.1)

and its Stratonovich counterpart dX_t = a(t, X_t) dt + M∑ j=1 b^j(t, X_t) ∘ dW^j_t (1.2)

where a is defined componentwise by a^i(t, X) = a^i(t, X) − 1 2 N∑ j=1 M∑ k=1 b^{j,k}(t, X) ∂b^{i,k} ∂x_j (t, X) for i = 1, . . ., N.

In many of the MAPLE procedures that follow we will use the parameters a1,. . ., aN to denote the drift coefficients and [b11, . . . , b1M],. . ., [bN1, . . . , bNM] to denote the diffusion coefficients.

NOTE: The indexing on the the bIJ here is the transpose of the usual matrix indexing; here the first index I corresponds to the column and the second J to the row due to the fact that b^{i,j} is the jth component of the diffusion coefficient vector b^j associated with the jth component W^j_t of the Wiener process.

2 Explicitly Solvable Scalar SDE

In this Section we consider general scalar SDE dX_t = a(t, X_t) dt + b(t, X_t) dW_t (2.1) that can be solved explicitly. First we look at the general linear scalar SDE dX_t = (a_1(t)X_t + a_2(t)) dt + (b_1(t)X_t + b_2(t)) dW_t (2.2) for which an explicit solution is always available. Then we consider nonlinear scalar SDE that can be reduced to linear scalar SDE and hence solved explicitly.

2.1 Linearsde routine

The general form of a scalar linear stochastic differential equation is dX_t = (a_1(t)X_t + a_2(t)) dt + (b_1(t)X_t + b_2(t)) dW_t (2.3) where the coefficients a_1, a_2, b_1, b_2 are specified functions of time t or constants. When b_1(t) ≡ 0 in (2.1) the SDE is the additive noise and when b_1(t) ≠ 0 it has multiplicative noise.

In the general case the SDE can be solved with the integrating factor Φ_{t,t0} = exp (∫ t t_0 ( a_1(s) − 1 2 b^2_1(s) ) ds + ∫ t t_0 b_1(s) dW_s ) , (2.4) and has the eplicit solution

X_t = Φ_{t,t0} ( X_{t0} + ∫ t t_0

(a_2(s) − b_1(s)b_2(s)) Φ^{−1}_{s,t0} ds + ∫ t t_0

b_2(s)Φ^{−1}_{s,t0} dW_s ) . (2.5) In the additive noise case, the SDE reduces to dX_t = (a_1(t)X_t + a_2(t)) dt + b_2(t) dW_t, (2.6) and the integrating factor to Φ_{t,t0} = exp (∫ t t_0 a_1(s) ds ) , in which case the explicit solution is X_t = Φ_{t,t0} ( X_{t0} + ∫ t t_0

a_2(s)Φ^{−1}_{s,t0} ds + ∫ t t_0

b_2(s)Φ^{−1}_{s,t0} dW_s ) . (2.7)

The routine stochastic[linearsde] returns the explicit solution of an SDE (2.1) with linear drift coefficient a(t, x) = α(t)x or α(t) and linear diffusion coefficient b(t, x) = β(t)x or β(t). Its CALLING SEQUENCE is

linearsde(a,b): with PARAMETERS a -- algebraic, given in the variables x and t
b -- algebraic, given in the variables x and t

representing the drift and diffusion coefficients of the SDE, the particular cases of which are identified by the routine. A suitable error message is returned if the coefficients of a nonlinear SDE are used.

stochastic[linearsde]:=proc(a::algebraic,b::algebraic)
local temp1,alpha,beta,gamma,delta,fundsoln,fundsoln2,soln,
default1,default2,default3;
   if diff(a,x,x) <> 0 or diff(b,x,x) <> 0 then
   ERROR(‘SDE not linear, try a reducible procedure‘)
   else
       alpha := diff(a,x);
       alpha := subs(t = s,alpha);
       beta := diff(b,x);
       beta := subs(t = s,beta);
       if diff(beta,s) = 0 then temp1 := beta*W;
       else temp1:=Int(beta,W = 0 .. t);
       fi;
       gamma := coeff(a,x,0);
       gamma := subs(t = s,gamma);
       delta := coeff(b,x,0);
       delta := subs(t = s,delta);
       fundsoln := exp(int(alpha-1/2*beta^2,s = 0 .. t)+temp1);
       fundsoln2 := subs(t = s,fundsoln);
       if beta = 0 then
           soln := fundsoln*(X[0]+int(1/fundsoln2*(gamma-beta*delta),
           s = 0 .. t)+Int(1/fundsoln2*delta,W = 0 .. t))
       else soln := fundsoln*(X[0]+Int(1/fundsoln2*(gamma-beta*delta),
       s = 0 .. t)+Int(1/fundsoln2*delta,W = 0 .. t))
       fi;
       default1 := Int(0,W = 0 .. t) = 0;
       default2 := Int(0,W = 0 .. s) = 0;
       default3 := Int(0,s = 0 .. t) = 0;
       soln := X[t] = subs(default1,default2,default3,soln);
   fi;
end:

The call linearsde(a,b); returns the explicit solution for a linear SDE (2.3) with drift coefficient a and diffusion coefficient b. The output consists of the variables X[t], X[0] and W , where X[t] denotes the explicit solution, X[0] the initial value of the solution, W a standard Wiener process, and t time.

The routine is used by the routine stochastic[explicit] and, in general, is not intended to be used on its own. It is part of the stochastic package and is usually loaded via the call with(stochastic), but can also be invoked via the call stochastic[linearsde].

EXAMPLE:Consider the scalar linear SDE with aditive noise dX_t = −X_t dt + 2 dW_t with drift a(t, x) = −x and diffusion coefficient b(t, x) = 2

>linearsde(-x,2);

X_t = e^{(−t)} ( X_0 + ∫ t 0 2 e^{(−s)} dW )

SEE ALSO: stochastic, stochastic[reducible], stochastic[explicit].

2.2 Reducible routine

A nonlinear scalar SDE dX_t = a(X_t) dt + b(X_t) dW_t (2.8) can be reduced to a linear scalar stochastic differential dY_t = dW_t (2.9) by a substitution y = h(x) = ∫ x ds b(s) , (2.10) giving the solution X_t = h^{−1}(W_t + h(X_0)), provided the drift has the form a(x) = 1 2 b(x)b^′(x). Here x = h^{−1}(y) is the inverse function of the function y = h(x).

More generally, if the drift has the form a(x) = α b(x)h(x) + 1 2 b(x)b^′(x) then the SDE can be reduced to the Langevin equation dY_t = αY_t dt + β dW_t giving the solution X_t = h^{−1} ( e^{αt}h(X_0) + e^{αt} ∫ t 0 e^{−αs} dW_s ) .

The routine stochastic[reducible] returns the explicit solution of a reducible SDE 2.8. Its CALLING SEQUENCE is

reducible(a,b); with PARAMETERS a - algebraic, given in the variables x and t.
b - algebraic, given in the variables x and t.

representing the drift and diffusion coefficient of the SDE (which actually should not depend explicitly on the t variable here). If the SDE is not of the above reducible form, then a suitable error message is returned.

stochastic[reducible]:=proc(a::algebraic,b::algebraic)
local beta,temp1,h,temp3,alpha,soln,soln1;
    h := int(1/b,x);
    temp1 := alpha*b*h+1/2*b*simplify(diff(b,x));
    temp1 = a;
    alpha := simplify(solve(",alpha));
    beta := alpha*h;
    if diff(alpha,x) = 0 then
        if alpha=0 then
            soln:=h=subs(x=X[0],h)+W;
            X[t]=simplify(solve(soln,x));
        else
        soln1 := h = exp(alpha*t)*subs(
                      x = X[0],h)+exp(alpha*t)*Int(exp(-alpha
                                      *s),W = 0 .. t);
        X[t] = solve(soln1,x);
        fi
    elif diff(beta,x) = 0 then
    X[t]=simplify(solve(h = beta*t+W+subs(x = X[0],h),x));
    else ERROR(‘non-linear SDE not reducible‘)
    fi
end:

The call reducible(a,b); returns the explicit solution for a reducible SDE with drift a and diffusion coefficient b. The output consists of the variables X[t], X[0] and W. where X[t] denotes the explicit solution, X[0] the initial value of the solution, W a standard Wiener process and t time.

This routine is used by the routine stochastic[explicit] and, in general, is not intended to be used on its own. It is part of the stochastic package and is usually loaded via the call with(stochastic), but can also be called via the call stochastic[reducible].

EXAMPLE: Consider the scalar nonlinear SDE

dX_t = 1 dt + 2 √ X_t dW_t.

>reducible(1,2*sqrt(x));

                                    1/2 2
                       X[t] = 2 X[0] W + W + X[0]

This SDE is reducible and the required solution is

X_t = X_0 + 2 √ X_0 W_t + (W_t)^2 .

SEE ALSO: stochastic, stochastic[linearsde], stochastic[explicit]

2.3 Explicit routine

The routine stochastic[explicit] applies the routines stochastic[linearsde] and stochastic[reducible] to a general scalar SDE and returns the explicit solution if the SDE is either linear or reducible as in the preceeding subsections. Its CALLING SEQUENCE is

explicit(a,b);

with PARAMETERS

a - algebraic, given in the variables x and t
b - algebraic, given in the variables x and t

representing the drift and the diffusion coefficient of the SDE. A suitable error message is returned if the SDE is not linear or reducibale to a linear SDE.

stochastic[explicit]:= proc(a::algebraic,b::algebraic)
if diff(a,x,x) = 0 and diff(b,x,x) = 0
then linearsde (a,b) else reducible(a,b)
fi
end:

The call explicit(a,b); returns the explicit solution for a scalar SDE with drift coefficient a and diffusion coefficient b of the appropriate form. The output consists of the variables X[t], X[0] and W , where X[t] denotes the explicit solution, X[0] the initial value, W denotes a standard Wiener process, and t time. The user is returned with a suitable error message if no known explicit solution exists.

This routine is part of the stochastic package and is usually loaded via the call with(stochastic). It can also be invoked via the call stochastic[explicit]. EXAMPLE: Consider the scalar SDE dX_t = 1 2 a^2X_t dt + aX_t dW_t, with drift a(x) =^1_2 ax and diffusion coefficient b(x) = ax, where a is a constant.

>explicit(1/2*a^2*x,a*x);

               X[t]=exp(aW)X[0]

This SDE is linear and thus explicitly solvable with the solution

X_t = X_0e^{aWt} .

SEE ALSO: stochastic, stochastic[linearsde], stochastic[reducible]

3 Ito Stochastic Calculus

We now consider a general N-dimensional Ito SDE dX_t = a(t, X_t) dt + M∑ j=1 b^j(t, X_t) dW^j_t (3.1)

with an M-dimensional Wiener process W_t = (W^1_t , · · · , W^M_t ). The operators L^0 and L6j with respect to this SDE which are defined by

L^0 = ∂ ∂t + N∑ k=1 a^k ∂ ∂x^k + 1 2 N∑ k,l=1 M∑ j=1 b^{k,j}b^{l,j} ∂^2 ∂x^k∂x^l (3.2) and L^j = N∑ k=1 b^{k,j} ∂ ∂x^k , j = 1, . . . , M (3.3) play a fundamental role in stochastic calculus through the Ito formula, stochastic Taylor expansions and numerical schemes for the SDE that are based on stochastic Taylor expansions.

3.1 LJ Operator routine

The routine stochastic[LJ] applies the partial differential operator L^j defined by 3.3. Its CALLING SEQUENCE

LJ(X,[[b11,..,b1M], . . ., [bN1,..,bNM]],j);

has PARAMETERS

                           X - algebraic, given in the variables x[N] and t,
[b11,..,b1M],..,[bN1,..,bNM] - lists of algebraics, given in the variables
                               x[N] and t,
                           j - integer,

where the bIJ are the components of the diffusion coefficient vectors of the SDE (3.1).

stochastic[LJ]:= proc(X::algebraic,b::list(list(algebraic)),j::integer)
sum(’op(j,op(k,b))*diff(X,x[k])’,’k’ = 1 .. nops(b))
end:

The call LJ(X,[[b11,..,b1M],..,[bN1,..,bNM]],j);] applies the partial differential operator L^j to the function X, where [b11, . . . , b1M], . . ., [bN1, . . . , bNM] denotes the components of the M N–dimensional diffusion coefficient vectors, where M is the dimension of the Wiener process and N is the dimension of the SDE (3.1), and j = 1, . . ., M denotes the “current” component of the Wiener process. The output variables are consistent with the variables used as input.

This routine is used by the routines

stochastic[Euler], stochastic[Milstein], stochastic[Taylor1.5], stochastic[Taylor2], stochastic[wkeuler], stochastic[wktay2], stochastic[itoformula], stochastic[MLJ].

In general, it is not intended for use on its own, but is part of the stochastic package and is usually loaded via the call with(stochastic). It can also be invoked via the call stochastic[LJ] EXAMPLE: Consider the function X(t, x_1, x_2) = x_2 and a 2–dimensional SDE driven by a 2–dimensional Wiener process with diffusion coefficient vectors

b^1 = b^{1,1} b^{1,2} = r 0 , b^2 = b^{2,1} b^{2,2} = 0 r , where r is a constant.

>LJ(x[2],[[r,0],[0,r]],2);

               r

>LJ(x[2],[[r,0],[0,r]],1);

               0

This has evaluated L^2X(t, x_1, x_2) = r and L^1X(t, x_1, x_2) = 0 SEE ALSO:

stochastic[L0], stochastic[MLJ], stochastic[SL0], stochastic[Euler], stochastic[Milstein], stochastic[Taylor1hlf], stochastic[Taylor2], stochastic[wkeuler], stochastic[wktay2], stochastic.

3.2 L0 Operator routine

The routine stochastic[L0] applies the partial differential operator L^0 to the function X. Its CALLING SEQUENCE

L0(X,[a1,..,aN], [[b11,..,b1M],.., [bN1,..,bNM]]);

has PARAMETERS

                           X - algebraic, given in the variables x[N] and t,
                    a1,..,aN - algebraics, given in the variables x[N] and t,
[b11,..,b1M],..,[bN1,..,bNM] - lists of algebraics, given in the variables
                                x[N] and t,

where a1, . . . , aN denotes the drift coefficients of the N–dimensional SDE (3.1), while [b11, . . . , b1M], . . ., [bN1, . . . , bNM] denote the components of the corresponding M N–dimensional diffusion coefficient vectors.

stochastic[L0]:=proc(X::algebraic,a::list(algebraic),
                                 b::list(list(algebraic)))
local part1,part2,part3;
    part1 := diff(X,t);
    part2 := sum(’a[k]*diff(X,x[k])’,’k’ = 1 .. nops(a));
    part3 := 1/2*sum(
    ’sum(’sum(’op(j,op(k,b))*op(j,op(l,b))*diff(X,x[k],x[l])’,
    ’j’ = 1 .. nops(op(1,b)))’,’k’ = 1 .. nops(a))’,’l’ = 1 .. nops(a)
    );
    part1+part2+part3;
end:

The call L0(X,[a1,..,aN],[[b11,..,b1M],..,[bN1,..,bNM]]); applies the partial differential operator L^0 to the function X. It is used by the routines

stochastic[Euler],stochastic[Milstein], stochastic[Taylor1.5], stochastic[wkeuler], stochastic[wktay2],stochastic[itoformula], stochastic[LFP], stochastic[MLJ].

In general, it is not intended for use on its own. The routine is part of the stochastic package and is usually loaded via the call with(stochastic) and can also be invoked via the call stochastic[L0] .

EXAMPLE: Consider the 2–dimensional SDE driven by a 2–dimensional Wiener process given by d X^1_t X^2_t = X^1_t X^2_t dt + r 0 dW^1_t + 0 r dW^2_t , where r is a constant, that is with drift with components a^1 = x_1, a^2 = x_2 and the constant diffusion coefficient vectors

b^1 = b^{1,1} b^{1,2} = r 0 , b^2 = b^{2,1} b^{2,2} = 0 r . Apply the corresponding operator L^0 to the function X(t, x_1, x_2) = x_2.

>L0(x[2],[x[1],x[2]],[[r,0],[0,r]]);

              x[2]

The result is L^0X(t, x_1, x_2) = x_2.

3.3 LFP Operator: Fokker–Planck Equation

The transition probabilities of the Ito stochastic differential equation (3.1) have densities p(s, x; t, y) which satisfy the Fokker-Planck equation, ∂p ∂t + N∑ i=1 ∂ ∂y_i { a^i(t, y)p } − 1 2 N∑ i,j=1 M∑ k=1 ∂^2 ∂y_i∂y_j { b^{i,k}(t, y)b^{j,k}(t, y)p } = 0 (3.4)

(s, x in p fixed) with the initial condition lim_{t↓s} p(s, x; t, y) = δ(x − y), where δ is the Dirac delta function on ℜ^N

We define the corresponding differential operator L^∗ as , which is in fact the adjoint of the L or L^0 operator above as L^∗p = ∂p ∂t + N∑ i=1 ∂ ∂y_i { a^i(t, y)p } − 1 2 N∑ i,j=1 M∑ k=1 ∂^2 ∂y_i∂y_j { b^{i,k}(t, y)b^{j,k}(t, y)p } . In MAPLE it will be called the LF P operator.

The CALLING SEQUENCE

LFP(P,[a1,..,aN], [[b11,..,b1M],.., [bN1,..,bNM]]);

has PARAMETERS

                           P - algebraic, given in the variables y[N] and t,
                    a1,..,aN - algebraics, given in the variables y[N] and t,
[b11,..,b1M],..,[bN1,..,bNM] - lists of algebraics, given in the variables
                                y[N] and t,

where a1, . . . , aN denotes the drift coefficients of the N–dimensional SDE (3.1), while [b11, . . . , b1M], . . ., [bN1, . . . , bNM] denote the components of the corresponding M N–dimensional diffusion coefficient vectors. The procedure has following code:

LFP:=proc(P::algebraic,a::list(algebraic),b::list(list(algebraic)))
local part1,part2,part3;
  part1 := diff(P,t);
  part2 := sum(’diff(a[k]*P,y[k])’,’k’ = 1 .. nops(a));
  part3 := 1/2*sum(
  ’sum(’sum(’diff(op(j,op(k,b))*op(j,op(l,b))*P,y[k],y[l])’,
  ’j’ = 1 .. nops(op(1,b)))’,’k’ = 1 .. nops(a))’,’l’ = 1 .. nops(a)
  );
  part1+part2-part3;
end:

EXAMPLE: Consider the Ito equation dX_t = a dt + b dW_t, where a and b are constant. If we want obtain the equation (3.4) for this Ito-equation we call

> LFP(p(s,x,t,y[1],[a],[[b]]);

and obtain ( ∂ ∂t p(s, x, t, y_1)) + a ( ∂ ∂y_1 p(s, x, t, y_1)) − 1 2 b^2 (

∂^2 ∂y_1^2 p(s, x, t, y_1)) = 0.

If we have a density function p(s, x, t, y) = 1 √ 2πb^2(t − s) e �(y�x�a�(t�s))^2 2b2(t�s) ).

and want know, if p(s, x, t, y) satisfy the Fokker–Planck equation, we call:

> LFP(p(s,x,t,y[1]),[a],[[b]]);.

The output of LFP is:

1 4 √ 2 %2 π b^2 (π b^2 (t − s))^{3/2} + 1 2 √ 2 ( %1 a b^2 (t − s) + 1 2 %1^2 b^2 (t − s)^2

) %2 √ π b^2 (t − s) − 1 2 a √ 2 %1 %2 √ π b^2 (t − s) b^2 (t − s) + 1 4 √ 2 %2 √ π b^2 (t − s) (t − s) − 1 4 √ 2 %1^2 %2 √ π b^2 (t − s) (t − s)^2 b^2 %1 := y_1 − x − a (t − s)

%2 := e^{(−1/2 %1} 2 b2 (t�s) ).

After the MAPLE command

simplify(%);

we obtain 0. Hence p(s, x, t, y) satisfies the Fokker–Planck equation.

3.4 Application of the partial differential operators

The routine stochastic[MLJ] applies one of the partial differential operators, L0 or LJ to the mapping X. Its CALLING SEQUENCE is

MLJ(X,[a1,..,aN],[[b11,..,b1M],..,[bN1,..,bNM]],j);

with PARAMETERS

                           X - algebraic, given in the variables x[N] and t.
                    a1,..,aN - algebraics, given in the variables x[N] and t.
[b11,..,b1M],..,[bN1,..,bNM] - lists of algebraics, given in
                                variables x[N] and t.
                           j - integer.

where a1, . . . , aN denotes the drift coefficients of the N–dimensional SDE (3.1), while the [b11, . . . , b1M], . . ., [bN1, . . . , bNM] denote the corresponding M N–dimensional diffusion coefficient vectors of the SDE (3.1).

stochastic[MLJ]:=
proc(X::algebraic,a::list(algebraic),b::list(list(algebraic)),j::integer)
local flag;
    flag := 0;
    if j = 0 then flag := L0(X,a,b) fi;
    if flag = 0 then flag := sum(’op(j,op(k,b))*diff(X,x[k])’,
    ’k’ = 1 .. nops(b)) fi;
    RETURN(flag)
end:

The call MLJ(X,[a1,..,aN],[[b11,..,b1M],..,[bN1,..,bNM]],j); computes the application of either the operator LJ or the operator L0 to the function X. Here a1, . . ., aN are the drift coefficients of the N–dimensional SDE and [b11, . . . , b1M], . . ., [bN1, . . . , bNM] are the M N–dimensional diffusion coefficient vectors, where M is the dimension of the Wiener process, while j = 1,..,M refers to the ‘current’ component of the Wiener process. The output variables are consistent with the variables used as input.

This routine is used by the routine stochastic[wktay3] and in general, is not intended for use on its own. It is part of the stochastic package and is usually loaded via the call with(stochastic), but can also be invoked directly via the call stochastic[MLJ].

EXAMPLES: Consider the 2–dimensional SDE driven by a 2–dimensional Wiener process given by d X^1_t X^2_t = X^2_t X^1_t dt + r 0 dW^1_t + 0 r dW^2_t , where r and s are constants, that is with drift with components a^1 = x_2, a^2 = x_1 and the constant diffusion coefficient vectors

b^1 = b^{1,1} b^{1,2} = r s , b^2 = b^{2,1} b^{2,2} = s r .

Apply the corresponding operators L^2 and L^0 to the function X(t, x_1, x_2) = x_2

>MLJ(x[2],[x[2],x[1]],[[r,s],[s,r]],2);

                r

>MLJ(x[2],[x[2],x[1]],[[r,s],[s,r]],0);

                x[1]

The result is L^2X(t, x_1, x_2) = r and L^0X(t, x_1, x_2) = x_1.

SEE ALSO: stochastic, stochastic[L0], stochastic[LJ], stochastic[SL0], stochastic[wktay3].

3.5 The Ito Formula

For a sufficiently smooth transformation U : [0, T ]×ℜ^N → ℜ of the solution X_t of the Ito SDE dX_t = a(t, X_t) dt + M∑ j=1 b^j(t, X_t) dW^j_t the scalar process Y_t = U(t, X_t) satissfies the a vector stochastic differential dY_t = ∂U ∂t + N∑ i=1 a^i ∂U ∂x_i + 1 2 N∑ i,j=1 M∑ l=1 b^{i,l}b^{j,l} ∂^2U ∂x_i∂x_j dt (3.5)

+ M∑ l=1 N∑ i=1 b^{i,l} ∂U^p ∂x_i dW^l_t where the terms are all evaluated at (t, X_t). This is called the Ito Formula.

In operator form it is dY_t = L^0U(t, X)t) dt + M∑ j=1 L^jU(t, X_t) dW^j_t .

itoformula:=proc(U::list(algebraic),a::list(algebraic),
                 b::list(list(algebraic)))
  local i,k,l0,lj,soln;
  for i from 1 to nops(U) do
    l0:=L0(U[i],a,b)*dt;
    lj:=0;

    for k from 1 to nops(b) do
      lj:=lj+LJ(U[i],b,k)*dW.i;
    od;
    soln[i]:=dX.i=l0 +lj;
  od;
  RETURN(eval(soln));
end:

EXAMPLE: Consider the function Y_t = U(t, X_t) = X^2_t where X_t is a solution of the Ito SDE dX_t = aX_t dt + bX_t dW_t Then we have:

> itoformula([(x[1])^2],[a],[[b]]);

  table([
                                2
        1 = (dX1 = (2 a x[1] + b ) dt + 2 b x[1] dW1)

        ]),

that is dY_t = (2aY_t + b^2) dt + 2bY_t dW_t.

3.6 Coloured Noise

We convert the N-dimensional Ito SDE (3.1) with a single white noise (i.e. m = 1) into its counterpart with coloured noise, that is driven by an Ornstein–Uhlenbeck or exponentially correlated coloured noise process. The resulting coloured noise equation is the (N + 1)-dimensional Ito SDE with scalar additive noise

dX_t = (a(t, X_t) + b(t, X_t) Z_t) dt (3.6) dZ_t = −γZ_t dt + β dW_t (3.7) The routine stochastic[colour] converts the SDE 3.1 with scalar white noise into its coloured noise counterpart (3.6)-(3.7). Its CALLING SEQUENCE is

colour([a1,..,aN],[b1,..,bN]);

with PARAMETERS

a1,..,aN - algebraics, given in the variables x[N] and t.
b1,..,bN - algebraics, given in the variables x[N] and t.

where a1, . . ., aN denotes the drift coefficients and b1, . . ., bN the diffusion coefficients of N–dimensional SDE with scalar noise.

stochastic[colour]:=proc(a::list(algebraic),b::list(algebraic))
local temp1,i;
    for i to nops(a) do temp1[i] := dx[i][t] = a[i]*dt+b[i]*z[t]*dt od;
    temp1[i] := dz[t] = -gamma*z[t]*dt+beta*dW[t];
    RETURN(eval(temp1))
end:

The call colour([a1,..,aN],[b1,..,bN]); converts N–dimensional SDE (3.1) with scalar white noise into its coloured noise form (3.6)-(3.7). The output consists of the variables z, x[N], W , gamma, beta and t. Here z denotes the Ornstein–Uhlenbeck process, (x[N], z) the state variable of the (N + 1)–dimensional SDE (3.6)-(3.7) and W a standard Wiener process, while gamma and beta denote parameters, usually provided from experimental data, and t denotes time.

This routine is part of the stochastic package and is usually loaded via the call with(stochastic). It can also be invoked directly via the call stochastic[colour].

EXAMPLE: Find the coloured noise counterpart of the 2–dimensional SDE with scalar noise

dX^1_t = X^2_t dt, dX^1_t = ( X^1_t ( α − ( X^1_t )2) − X^2_t ) dt + σ dW_t.

>colour([a(x[1],t)],[b(x[1],t)]);

table([
               1 = (dx[1][t] = a(x[1], t) dt + b(x[1], t) z[t] dt)

               2 = (dz[t] = - gamma z[t] dt + beta dW[t])
     ])

>colour([x[2],x[1]*(alpha-x[1]^2)-x[2]],[0,sigma*x[1]]);

table([
        1 = (dx[1][t] = x[2] dt)

                                           2
        2 = (dx[2][t] = (x[1] (alpha - x[1] ) - x[2]) dt + sigma x[1] z[t] dt)

        3 = (dz[t] = - gamma z[t] dt + beta dW[t])
     ])

The resulting coloured noise system is

dX^1_t = X^2_t dt

dX^1_t = ( X^1_t ( α − ( X^1_t )2) − X^2_t + σ Z_t ) dt dZ_t = −γ Z_t dt + β dW_t SEE ALSO: stochastic

4 Stratonovich Stochastic Caculus

We consider a the N-dimensional Ito SDE with an M-dimensional Wiener process dX_t = a(t, X_t) dt + M∑ j=1 b^j(t, X_t) dW^j_t (4.1)

and its Stratonovich counterpart dX_t = a(t, X_t) dt + M∑ j=1 b^j(t, X_t) ∘ dW^j_t . (4.2)

Here a given a is defined componentwise by a^i(t, X) = a^i(t, X) − 1 2 N∑ j=1 M∑ k=1 b^{j,k}(t, X) ∂b^{i,k} ∂x_j (t, X) for i = 1, . . ., N, while a given a is defined componentwise by a^i(t, X) = a^i(t, X) + 1 2 N∑ j=1 M∑ k=1 b^{j,k}(t, X) ∂b^{i,k} ∂x_j (t, X) for i = 1, . . ., N.

4.1 Ito–Stratonovich correction

Here we introduce a procedure which applies the drift-correction formula to convert the Ito drift coefficient a into the corresponding Stratonovich drift coefficient a. This is the routine stochastic[correct] which has CALLING SEQUENCE:

correct([a1,..,aN],[[b11,..,b1M],..,[bN1,..,bNM]],i);

with PARAMETERS

                   a1,..,aN - algebraics, given in the variables x[N] and t.
b11,..,b1M],..,[bN1,..,bNM] - lists of algebraics, given in the variables
                              x[N] and t.
                          i - integer.

where a1, . . . , aN denote the drift coefficients of the N–dimensional SDE (4.2) and [b11, . . . , b1M], . . ., [bN1, . . . , bNM] denote the components of the corresponding M N–dimensional diffusion coefficient vectors, M being the dimension of the Wiener process W . The index i = 1, . . ., N denotes the ‘current’ component of the SDE. The output variables are consistent with the variables used as input.

stochastic[correct]:=proc(a::list(algebraic),b::list(list(algebraic)),i) a[i]-
 1/2*sum(’LJ(op(j,op(i,b)),b,j)’,’j’ = 1 .. nops(op(1,b)));
end:

The call correct([a1,..,aN],[[b11,..,b1M],..,[bN1,..,bNM]],i); converts the drift coefficient of the Ito SDE (4.1) into that of its Stratonovich form (4.2).

This routine is used by the routine stochastic[Taylor2]. It is part of the stochastic package and is usually loaded via the call with(stochastic), but can also be invoked via the call stochastic[correct].

EXAMPLES: Both examples have N = M = 2 with the same Ito drift vector a^1(t, x_1, x_2) = x_1, a^2(t, x_1, x_2) = x_2, and the constant diffusion coefficient vectors

b^1 = b^{1,1} b^{1,2} = r 0 , b^2 = b^{2,1} b^{2,2} = 0 r , where r is a constant, in the first case and the variable diffusion coefficient vectors

b^1 = b^{1,1} b^{1,2} = x_1 0 , b^2 = b^{2,1} b^{2,2} = 0 r , in the second case.

>correct([x[1],x[2]],[[r,0],[0,r]],2);

               x[2]

>correct([x[1],x[2]],[[x[1],0],[0,r]],1);

               1/2 x[1]

The application of the routine here produces the result a^1(t, x_1, x_2) = x_2 in the first case and a^2(t, x_1, x_2) =^1_2 x_1 in the second case. SEE ALSO: stochastic, stochastic[Taylor2]

4.2 Ito–Stratonovich correction: both directions

The next procedure combines the Ito to Stratonovich conversion with the Stratonovich to Ito conversion procedure of the last subsection.

conv:=proc(a::list(algebraic),b::list(list(algebraic)),c::algebraic)
local temp,i;
if c=ito then
      for i from 1 to nops(a) do temp[i]:=op(i,a)-1/2*sum(’sum(’op(k,op(j,b))
          *diff(op(k,op(i,b)),x[j])’,’k’=1..nops(op(1,b)))’,’j’=1..nops(a));
     od;

elif c=strat then
       for i from 1 to nops(a) do
temp[i]:=op(i,a)+1/2*sum(’sum(’op(k,op(j,b))
           *diff(op(k,op(i,b)),x[j])’,’k’=1..nops(op(1,b)))’,’j’=1..nops(a));
     od;
else
     ERROR(‘Must enter either ito or strat for the 3rd argument‘)
fi;
RETURN(map(simplify,eval(temp)))
end:

EXAMPLE: Consider the Ito SDE

dX_t = −a^2X_t(1 − X^2_t ) dt + a(1 − X^2_t ) dW_t

To obtain the Stratonovich SDE use

> conv([-a^2*x[1](1-x[1]^2)],[[a*(1-x[1]^2)]],ito);

and obtain

table([ 1 = 0 ]), which means that the desired Stratonovich SDE is dX_t = 0 dt − a(1 − X^2_t ) ∘ dW_t = −a(1 − X^2_t ) ∘ dW_t.

The other direction gives the original Ito SDE back.

> conv([0],[[a*(1-x[1]^2)]],strat);

table([

1 = a^2 x_1 (−1 + x_1^2) ]). 4.3 Stratonovich L0 operator

The L^0 operator of of Ito calculus needs to be changed in Stratonovich calculus to

L^0 = ∂ ∂t + N∑ k=1 a^k ∂ ∂x^k , (4.3) while the L^j operator of of Ito calculus remains unchanged in Stratonovich calculus. The Stratonovich operator L^0 applied to a function X is produced by the routine stochastic[SL0] with CALLING SEQUENCE:

SL0(X,[a1,..,aN],[[b11,..,b1M],..,[bN1,..,bNM]]);

which has PARAMETERS:

                           X - algebraic, given in the variables x[N] and t.
                    a1,..,aN - algebraics, given in the variables x[N] and t.
[b11,..,b1M],..,[bN1,..,bNM] - lists of algebraics, given in the variables
                               x[N] and t.

where a1, . . . , aN denote the drift coefficients of the N–dimensional SDE (4.2) and [b11, . . . , b1M], . . ., [bN1, . . . , bNM] denote the components of the corresponding M N–dimensional diffusion coefficient vectors, M being the dimension of the Wiener process W . The output variables are consistent with the variables used as input.

stochastic[SL0]:=proc(X::algebraic,a::list(algebraic),b::list(list(algebraic)))
local part1,part2;
    part1 := diff(X,t); part2 := sum(’a[k]*diff(X,x[k])’,’k’ = 1 .. nops(a));
    part1+part2;
end:

The call SL0(X,[a1,..,aN],[[b11,..,b1M],..,[bN1,..,bNM]]); computes the application of the Stratonovich version of the operator L0 to X.

This routine is used by the routine stochastic[Taylor2]. In general, it is not intended for use on its own. It is part of the stochastic package and is usually loaded via the call with(stochastic), but can also be invoked directly via the call stochastic[SL0]. EXAMPLE: L^0X is computed for the function X(t, x_1, x_2) = x_2 and the 2–dimensional Stratonovich SDE with drift components a^1(t, x_1, x_2) = x_1, a^2(t, x_1, x_2) = x_2, and the constant diffusion coefficient vectors

b^1 = b^{1,1} b^{1,2} = r 0 , b^2 = b^{2,1} b^{2,2} = 0 r , where r is a constant.

>SL0(x[2],[x[1],x[2]],[[r,0],[0,r]]);

                x[2]

giving the result L^0X(t, x_1, x_2) = x_2.

SEE ALSO: stochastic, stochastic[L0], stochastic[LJ], stochastic[MLJ], stochastic[Taylor2]

4.4 Stratonovich transformation formula

For a sufficiently smooth transformation U : [0, T ]×ℜ^N → ℜ of the solution X_t of the Stratonovich SDE dX_t = a(t, X_t) dt + M∑ j=1 b^j(t, X_t) ∘ dW^j_t the scalar process Y_t = U(t, X_t) satisfies the a vector Stratonovich stochastic differential dY_t = ( ∂U ∂t + N∑ i=1 a^i ∂U ∂x_i ) dt + M∑ l=1 N∑ i=1 b^{i,l} ∂U ∂x_i ∘ dW^l_t (4.4)

where the terms are all evaluated at (t, X_t). In operator form this is dY_t = L^0U(t, X)t) dt + M∑ j=1 L^jU(t, X_t) ∘ dW^j_t .

chainrule:=proc(U::list(algebraic),a::list(algebraic),
                b::list(list(alge braic)))
  local i,k,l0,lj,soln;
  for i from 1 to nops(U) do
    l0:=SL0(U[i],a,b)*dt;
    lj:=0;
    for k from 1 to nops(b) do
      lj:=lj+LJ(U[i],b,k)*odW.i;
    od;
    soln[i]:=dX.i=l0 +lj;
  od;
  RETURN(eval(soln));
end:

EXAMPLE: Consider the function U(t, X_t) = X^2_t and the Stratonovich SDE dX_t = aX_t dt + bX_t ∘ dW_t. Then we have:

> chainrule([(x[1])^2],[a ],[[b]]);

  table([

        1 = (dX1 = 2 a x[1] dt + 2 b x[1] odW1)

        ]).

That means: dY_t = 2aY_t dt + 2bY_t ∘ dW_t.

5 Linear Vector SDEs

5.1 Linearization

We consider the linearization of an N-dimensional Ito SDE with an M-dimensional Wiener process dX_t = a(t, X_t) dt + M∑ j=1 b^j(t, X_t) dW^j_t (5.1)

about fixed solution X̄_t, resulting in the linear vector SDE dZ_t = A(t)Z_t dt + M∑ j=1 B^j(t)Z_t dW^j_t (5.2)

where A(t)^{i,j} = ∂a^i ∂x^j (t, X̄(t)), B^k(t)^{i,j} = ∂b^{k,i} ∂x^j (t, X̄(t)) for i, j = 1, . . ., N and k = 1, . . ., M.

The routine stochastic[linearize] has the CALLING SEQUENCE:

linearize( A,B,C);

with PARAMETERS

A - the list of $[a[1],a[2],...,a[N]]$,
B - listlist of $[[b[11],...,b[1M]],...,[b[N1],...,b[NM] ]],
C - list of [Xbar[1],...,Xbar[N]], where Xbar is the stationary solution
    of the SDE.

The procedure linearize has the following code:

linearize:=proc(a::list(algebraic),b::list(list(algebraic)),c::list(algebraic))
local i,tempA,tempB,j,k,l;
tempA:=array(1..nops(a),1..nops(a));
for i from 1 to nops(a) do for j from 1 to nops(a)
                        do tempA[i,j]:=diff(op(i,a),x[j]);
  od; od;
for i from 1 to nops(a) do for j from 1 to nops(a) do for l from 1 to nops(c)
                        do tempA[i,j]:=subs(x[l]=op(l,c),tempA[i,j]);
 od; od; od;
for k from 1 to nops(op(1,b)) do tempB[k]:=array(1..nops(a),1..nops(a));
for i from 1 to nops(a) do for j from 1 to nops(a)
                        do tempB[k][i,j]:=diff(op(k,op(i,b)),x[j]);
 od; od;
for i from 1 to nops(a) do for j from 1 to nops(a) do for l from 1 to nops(c)

                        do tempB[k][i,j]:=subs(x[l]=op(l,c),tempB[k][i,j]);
 od; od; od;

od;
RETURN(A=map(simplify,convert(eval(tempA),matrix)),B=eval(tempB))
end:

EXAMPLE: Consider the 2-dimensional Ito SDE

dX^1_t = X^2_t dt, dX^2_t = (−bX^2_t − sin X^1_t − c sin 2X^1_t )dt + (−a(X^2_t )^2 + sin X^1_t )dW_t, where a, b and c are constants and W_t is a scalar Wiener process.

a^2 = X^2(t) a^2 = −bX^2 − sin(X^1) − c sin(2X^1) − a^2(X^2)^3 + aX^2 sin(X^1).

> linearize([x[2],-b*x[2]-sin(x[1])-c*sin(2*x[1])],[[0],
  -a*(x[2])^2+sin(x[1])]],[0,0]);

Then A = [ 0 1 −1 − 2c −b ] and B = [ 0 0 1 0 ] , so the linearized SDE (5.2) is dZ^1_t = Z^2_t dt dZ^2_t = ( (−1 − 2c)Z^1_t − bZ^2_t ) dt + Z^1_t dW_t. 5.2 Second moment equation

We consider the N–dimensional linear Ito SDE dZ_t = A(t)Z_t dt + M∑ k=1 B^k(t)Z_t dW^k_t , (5.3)

where A, B^1, B^2, . . ., B^M are N ×N matrices.

The N×N matrix valued second moment P (t) = E(Z_tZ^⊤_t ) satisfies the deterministic matrix differential equation dP dt = A(t)P + P A(t)^⊤ + M∑ k=1 B^k(t)P B^k(t)^⊤, which is linear in P . On account of the symmetry of the matrix P , we can write this equation as a linear system of the from dp̃ dt = A(t)p̃ (5.4)

where p̃ is an^1_2 N(N + 1)-dimensional vector consisting of the free components of P and A(t) is a square matrix.

The routine stochastic[momenteqn] has the CALLING SEQUENCE:

momenteqn( A,B);

with PARAMETERS

A - the matrix A(t),
B - lists of matrixes B1(t),...,BM(t).

The procedure momenteqn calculates the new matrix A(t) and has the following code:

momenteqn:=proc(A,B)
  local i,j,k,N,Btmp,Ctmp;
  global Neues_A;
  if type(A,array) then Btmp:=convert(A,listlist);else Btmp:=A; fi;
  N:=nops(Btmp);

Neues_A:=array(1..N*(N+1)/2,1..N*(N+1)/2);Ctmp:=array(1..N*(N+1)/2,1..N*(N+1)/2)
;
  ap(A);
  pa(A);
  for i from 1 to N*(N+1)/2 do
   for j from 1 to N*(N+1)/2 do
     Ctmp[i,j]:=0;
   od;
  od;
  for k from 1 to nops(B) do
   bpb(B[k]);
   for i from 1 to N*(N+1)/2 do
     for j from 1 to N*(N+1)/2 do
       Ctmp[i,j]:=Ctmp[i,j]+B3[i,j];
     od;
   od;
  od;
  for i from 1 to N*(N+1)/2 do
   for j from 1 to N*(N+1)/2 do
     Neues_A[i,j]:=B1[i,j]+B2[i,j]+Ctmp[i,j];
   od;
  od;
  RETURN(evalm(Neues_A));
end:

This procedure requires other procedures. It is not necesary to call these procedures, but they are must be declared (in the package) before calling the procedure momenteqn (see the APPENDIX).

EXAMPLE: Let A = [ A_{1, 1} A_{1, 2} A_{2, 1} A_{2, 2} ] and B = [ B_{1, 1} B_{1, 2} B_{2, 1} B_{2, 2} ] .

We set E = [B] (list of arrays) and call momenteqn(A,E), obtaining

2 A_{1, 1} + B_{1, 1}^2 2 A_{1, 2} + 2 B_{1, 2} B_{1, 1} B_{1, 2}^2 A_{2, 1} + B_{1, 1} B_{2, 1} A_{1, 1} + A_{2, 2} + B_{1, 2} B_{2, 1} + B_{1, 1} B_{2, 2} A_{1, 2} + B_{1, 2} B_{2, 2} B_{2, 1}^2 2 A_{2, 1} + 2 B_{2, 2} B_{2, 1} 2 A_{2, 2} + B_{2, 2}^2 . The equation (5.4) is thus

dp̃_1 = ( 2 A_{1, 1} + B_{1, 1}^2 ) p̃_1 + (2 A_{1, 2} + 2 B_{1, 2} B_{1, 1}) p̃_2 + ( B_{1, 2}^2 ) p̃_3

dp̃_2 = (A_{2, 1} + B_{1, 1} B_{2, 1}) p̃_1 + (A_{1, 1} + A_{2, 2} + B_{1, 2} B_{2, 1} + B_{1, 1} B_{2, 2}) p̃_2 + (A_{1, 2} + B_{1, 2} B_{2, 2}) p̃_3

dp̃_3 = ( B_{2, 1}^2 ) p̃_1 + (2 A_{2, 1} + 2 B_{2, 2} B_{2, 1}) p̃_2 + ( 2 A_{2, 2} + B_{2, 2}^2 ) p̃_3

The procedures matrix2pvector and pvector2pmatrix transform a symmetry matrix to a vector and a vector to an symmetry matrix, respectively. Here they are used to change the matrix P to the vector p̃ and the vector p̃ to the matrix P , respectively. The procedure matrix2pvectorhas the following code (see the APPENDIX for the procedure pvector2pmatrix):

matrix2pvector:=proc(p)
  local i,j,k,ptmp;
  global pvector;
  if type(p,array) then ptmp:=convert(p,listlist);else ptmp:=p; fi;
  pvector:=array(1..nops(ptmp)*(nops(ptmp)+1)/2);
  k:=0;
  for i from 1 to nops(ptmp) do
    if (i>1) then k:=k+(nops(ptmp)-i+2); fi;
    for j from i to nops(ptmp) do
       pvector[k+j-i+1]:=ptmp[i,j];
    od;
  od;
  RETURN(eval(pvector));
end:

EXAMPLE: Let

P = 2 5 10 17 26 5 6 11 18 27 10 11 12 19 28 17 18 19 20 29 26 27 28 29 30 .

After calling pmatrix2pvector(P); we obtain

[2, 5, 10, 17, 26, 6, 11, 18, 27, 12, 19, 28, 20, 29, 30].

EXAMPLE: Consider the 2 × 2 symmetric matrix written in the listlist form P := [[1, 2], [2, 4]].

For p̃ the procedure pmatrix2pvector(P); yields

[1, 2, 4].

These examples for the first procedure show that it is possible to have a matrix or a listlist as an input parameter.

5.3 Spherical coordinates

We consider the N–dimensional linear Stratonovich SDE dZ_t = A(t)Z_t dt + M∑ k=1 B^k(t)Z_t ∘ dW^k_t , (5.5)

where A, B^1, B^2, . . ., B^M are N×N matrices, and convert it to spherical coordinates r = |z| and s = z/|z| ∈ S^{N−1} (assuming z ≠ 0). The resulting system of equations is

dR_t = R_t q^0(S_t) dt + M∑ k=1 R_t q^k(S_t) ∘ dW^k_t (5.6) dS_t = h(S_t, A) dt + M∑ k=1 h(S_t, B^k) ∘ dW^k_t (5.7)

where q(s) = s^⊤As + M∑ k=1 ( 1 2 s^⊤ ( B^k + ( B^k )⊤) s − ( s^⊤B^ks )2) , q^0(s) = s^⊤As, q^k(s) = s^⊤B^ks, h(s, A) = ( A − (s^⊤As)I ) s. The routine stochastic[sphere] has the CALLING SEQUENCE:

sphere(A,B);

with PARAMETERS

A - the matrix A(t),
B - lists of matrixes $B1(t),...,BM(t).

It calculates the values for q(s), q^0(s), q^k(s), h(s, A) and h(s, B^k) for k = 1, . . . , M and has the following code:

sphere:=proc(a,b)
  global q,q0,qk,h,hk;
  local i,j,k,tempa,tempb,stempbs,N,tmp;

  if type(a,array) then tmp:=convert(a,listlist);else tmp:=a; fi;
  N:=nops(tmp);
  hk:=evaln(hk); h:=evaln(h); qk:=evaln(qk);
  q:=evaln(q); q0:=evaln(q0);tempa:=evaln(tempa);tempb:=evaln(tempb);

  q0:=sum(’sum(’s[i]*a[i,j]’,’i’=1..N)*s[j]’,’j’=1..N);
  for k from 1 to nops(b) do
   qk[k]:=sum(’sum(’s[i]*b[k][i,j]’,’i’=1..N)*s[j]’,’j’=1..N);
  od;

  for k from 1 to nops(b) do

stempbs[k]:=sum(’sum(’s[i]*(b[k][i,j]+b[k][j,i])’,’i’=1..N)*s[j]’,’j’=1..N);
  od;

  q:=q0+ sum(’0.5*stempbs[k]-qk[k]^2’,’k’=1..nops(b));

  for i from 1 to N do
   for j from 1 to N do
     if (i=j) then tempa[i,i]:=a[i,i]-q0;
             else tempa[i,j]:=a[i,j];
     fi;
   od;
  od;

  for i from 1 to N do
   h[i]:=sum(’tempa[i,j]’,’j’=1..N);
  od;

  for k from 1 to nops(b) do
   for i from 1 to N do
     for j from 1 to N do
       if (i=j) then tempb[k][i,i]:=b[k][i,i]-qk[k];
               else tempb[k][i,j]:=b[k][i,j];
       fi;
     od;
   od;
  od;

  for k from 1 to nops(b) do
   for i from 1 to N do
     hk[k][i]:=sum(’tempb[k][i,j]’,’j’=1..N);
   od;
  od;

end:

The variables here are defined as global, which means that to determine the value of q(s), for example, we call sphere(A,B): q;.

EXAMPLE: Consider the matrices

A := [ 2 2 1 1 ] and B := [ 3 1 4 2 ]

which are input into Maple as:

>A:=array(1..2,1..2,[[2,2],[1,1]]):
>B1:=array(1..2,1..2,[[3,1],[4,2]]):
>B:=[B1]:

Then

> sphere(A,B):
> q;

(2 s_1 + s_2) s_1 + (2 s_1 + s_2) s_2 + .5000000000 (6. s_1 + 5. s_2) s_1 + .5000000000 (5. s_1 + 4. s_2) s_2 − 1. ((3. s_1 + 4. s_2) s_1 + (s_1 + 2. s_2) s_2)^2

> q0;

(2 s_1 + s_2) s_1 + (2 s_1 + s_2) s_2

> print(qk);

table([ 1 = (3 s_1 + 4 s_2) s_1 + (s_1 + 2 s_2) s_2 ])

> print(h);

table([ 1 = 4 − (2 s_1 + s_2) s_1 − (2 s_1 + s_2) s_2 2 = 2 − (2 s_1 + s_2) s_1 − (2 s_1 + s_2) s_2 ])

> print(hk);

table([ 1 = table([ 1 = 4 − (3 s_1 + 4 s_2) s_1 − (s_1 + 2 s_2) s_2 2 = 6 − (3 s_1 + 4 s_2) s_1 − (s_1 + 2 s_2) s_2 ]) ]) 6 Strong Numerical Schemes

We consider a the N-dimensional Ito SDE with an M-dimensional Wiener process dX_t = a(t, X_t) dt + M∑ j=1 b^j(t, X_t) dW^j_t (6.1)

and some strong stochastic Taylor schemes for this equation. The final order 2.0 scheme will be the corresponding Stratonovich SDE.

In all of the schemes that follow, the coefficients are all evaluated at the point (t_n, Y_n).

6.1 Euler scheme

The Euler scheme has the componentwise form

Y^k_{n+1} = Y^k_n + a^k Δ_n + M∑ j=1 b^{k,j} ΔW^j_n (6.2)

for k = 1, . . ., N, where Δ_n = t_{n+1} − t_n

is the length of the nth time step and

ΔW^j_n = W^j_{τn+1} − W^j_{τn}

is the N(0; Δ_n) distributed increment of the jth component of the M-dimensional standard Wiener process W on the discretization subinterval [τ_n, τ_{n+1}]. Here ΔW^{j1}_n and ΔW^{j2}_n are independent for j_1 ≠ j_2.

The routine stochastic[Euler] constructs stochastic Taylor scheme of strong order 0.5 known as the Euler scheme for an Ito SDE 6.1. Its CALLING SEQUENCE is Euler([a1,..,aN],[[b11,..,b1M],..,[bN1,..,bNM]]);

with PARAMETERS

                    a1,..,aN - algebraics, given in the variables x[N] and t.
[b11,..,b1M],..,[bN1,..,bNM] - lists of algebraics, given in variables x[N]
                               and t.

representing the drift and the diffusion coefficient vectors of the SDE.

stochastic[Euler]:=proc(a::list(algebraic),b::list(list(algebraic)))
local i,u,soln;
    for i to nops(a) do
        soln[i] := Y.i[n+1] = Y.i[n]+L0(x[i],a,b)*Delta[n]+sum(’LJ(x[i],b,j)*
        Delta*W.j[n]’,’j’ = 1 .. nops(op(1,b)));
        for u to nops(a) do soln[i] := subs(x[u] = Y.u[n],soln[i]) od
    od;
    RETURN(eval(soln))
end:

The call Euler([a1,..,aN],[[b11,..,b1M],..,[bN1,..,bNM]]); returns the Euler scheme for an N–dimensional Ito SDE with M–dimensional noise which has drift coefficients a1,. . ., aN and diffusion coefficients [b11, . . . , b1M],. . ., [bN1, . . . , bNM].

The output consists of the variables Y N[n], DeltaW M[n], and Delta[n]. Y N[n] denotes the strong order 0.5 stochastic Taylor approximation to x[N] at the n-th step. DeltaW M[n] denotes the change in the M–dimensional Wiener process at the n-th step. Delta[n] denotes the step size at the n-th step.

This routine is part of the stochastic package and is usually loaded via the call with(stochastic). It can also be invoked driectly via the call stochastic[Euler]. EXAMPLE: Consider the 2–dimensional SDE driven by a 2–dimensional Wiener process W_t = (W^1_t , W^2_t ), given by d X^1_t X^2_t = X^2_t X^1_t dt + r s dW^1_t + s r dW^2_t ,

that is with drift components a^1(t, x_1, x_2) = x_2, a^2(t, x_1, x_2) = x_1 and the constant diffusion coefficient vectors

b^1 = b^{1,1} b^{1,2} = r s , b^2 = b^{2,1} b^{2,2} = s r , where r and s are constants.

>Euler([x[2],x[1]],[[r,s],[s,r]]);

table([
    1 = (Y1[n + 1] = Y1[n] + Y2[n] Delta[n] + r Delta W1[n] + s Delta W2[n])

    2 = (Y2[n + 1] = Y2[n] + Y1[n] Delta[n] + s Delta W1[n] + r Delta W2[n])
     ])

The resulting Euler scheme is

Y^1_{n+1} Y^2_{n+1} = Y^1_n Y^2_n + Y^2_n Y^1_n

Δ_n + r s ΔW^1_n + s r ΔW^2_n .

SEE ALSO: stochastic, stochastic[L0], stochastic[LJ], stochastic[Milstein], stochastic[Taylor1hlf], stochastic[Taylor2] .

6.2 Milstein scheme

The Milstein scheme has the componentwise form

Y^k_{n+1} = Y^k_n + a^k Δ_n + M∑ j=1 b^{k,j}ΔW^j_n + M∑ j_1,j_2=1

L^{j1} b^{k,j2} I_{(j1,j2);n}, (6.3)

where k = 1, . . ., N and I_{(j1,j2);n} is the multiple Ito integral

I_{(j1,j2);n} = ∫ τ_{n+1} τn ∫ s_1 τn dW^{j1}_{s2} dW^{j2}_{s1} (6.4)

in general, with I_{(j1,j1);n} = 1 2 {( ΔW^{j1}_n )2 − Δ_n }

and similarly for I_{(j2,j2);n}.

The routine stochastic[Milstein] constructs stochastic Taylor scheme of strong order 1.0 known as the Milstein scheme for an Ito SDE (6.1). Its CALLING SE- QUENCE is

Milstein([a1,..,aN],[[b11,..,b1M],..,[bN1,..,bNM]]);

with PARAMETERS

                    a1,..,aN - algebraics, given in the variables x[N] and t.
[b11,..,b1M],..,[bN1,..,bNM] - lists of algebraics, given in variables x[N]
                               and t.

representing the driftand the diffusion coefficient vectors of the SDE.

stochastic[Milstein]:=proc(a::list(algebraic),b::list(list(algebraic)))
local u,i,soln;
    for i to nops(a) do
        soln[i] := Y.i[n+1] = Y.i[n]+L0(x[i],a,b)*Delta[n]+sum(’LJ(x[i],b,j)*
        Delta*W.j[n]’,’j’ = 1 .. nops(op(1,b)))+
            sum(’sum(’LJ(op(j2,op(i,b)),b,j1)*I[j1,j2]’,
            ’j1’ = 1 .. nops(op(1,b)))’,’j2’ = 1 .. nops(op(1,b)));
        for u to nops(a) do soln[i] := subs(x[u] = Y.u[n],soln[i]) od
    od;
    RETURN(eval(soln))
end:

The call Milstein([a1,..,aN],[[b11,..,b1M],..,[bN1,..,bNM]]); returns the Milstein scheme for an N–dimensional SDE (6.1) with M–dimensional noise which has drift coefficients a1, . . ., aN and diffusion coefficient vectors [b11, . . . , b1M], . . ., [bN1, . . . , bNM]. The output consists of the variables Y N[n], DeltaW M[n], Delta[n] and I[(j1, j2)]. Here Y N[n] denotes the strong order 1.0 stochastic Taylor approximation to x[N] at the n-th step, DeltaW M[n] denotes the increment in the M– dimensional Wiener process at the n-th step, Delta[n] denotes the step size at the n-th step, and I[(j1, j2)] denotes the double Ito integral (6.4)

This routine is part of the stochastic package and is usually loaded via the call with(stochastic). It can also be invoked directly via the call stochastic[Milstein].

EXAMPLE: Consider the 2–dimensional SDE driven by a 2–dimensional Wiener process W_t = (W^1_t , W^2_t ), given by d X^1_t

X^2_t = X^2_t

X^1_t dt + r s dW^1_t + s r dW^2_t , that is with drift components a^1(t, x_1, x_2) = x_2, a^2(t, x_1, x_2) = x_1 and the constant diffusion coefficient vectors

b^1 = b^{1,1} b^{1,2} = r s , b^2 = b^{2,1} b^{2,2} = s r , where r and s are constants.

>Milstein([x[2],x[1]],[[r,s],[s,r]]);

table([
    1 = (Y1[n + 1] = Y1[n] + Y2[n] Delta[n] + r Delta W1[n] + s Delta W2[n])

    2 = (Y2[n + 1] = Y2[n] + Y1[n] Delta[n] + s Delta W1[n] + r Delta W2[n])
      ])

The resulting Milstein scheme is

Y^1_{n+1} Y^2_{n+1} = Y^1_n Y^2_n + Y^2_n Y^1_n

Δ_n + r s ΔW^1_n + s r ΔW^2_n ,

which is actually the same as the Euler scheme in this case because the SDE here has additive noise.

SEE ALSO: stochastic, stochastic[L0], stochastic[LJ], stochastic[Euler], stochastic[Taylor1hlf], stochastic[Taylor2] .

6.3 Order 1.5 strong stochastic Taylor scheme

The kth component of the order 1.5 strong Taylor scheme is given by Y^k_{n+1} = Y^k_n + a^k Δ_n + 1 2 L^0a^k Δ^2_n (6.5) + M∑ j=1 ( b^{k,j} ΔW^j_n + L^0b^{k,j}I_{(0,j);n} + L^ja^k I_{(j,0);n} )

+ M∑ j_1,j_2=1

L^{j1} b^{k,j2} I_{(j1,j2);n} + M∑ j_1,j_2,j_3=1 L^{j1} L^{j2} b^{k,j3} I_{(j1,j2,j3);n} where I_{(j1,j2,j3);n} is the multiple Ito integral I_{(j1,j2,j3);n} = ∫ t_{n+1} tn ∫ s_1 τn ∫ s_2 τn dW^{j1}_{s3} dW^{j1}_{s2} dW^{j2}_{s1} (6.6) in general, with I_{(j1,j1,j1);n} = 1 2 { 1 3 ( ΔW^{j1}_n )2 − Δ_n } ΔW^{j1}_n and I_{(0,j);n} = ΔW^{j1}_n Δ_n − I_{(j,0);n} in specific cases. Here the random variable ΔZ^j_n := I_{(j,0);n} is N(0;^1_3 Δ_n) normally distributed and has covariance E(ΔZ^j_nΔW^j_n =^1_2 Δ^2_n.

The routine stochastic[Taylor1hlf] constructs stochastic Taylor scheme of strong order 1.0 Taylor scheme for an Ito SDE (6.1). Its CALLING SEQUENCE is

Taylor1hlf([a1,..,aN],[[b11,..,b1M],..,[bN1,..,bNM]]);

with PARAMETERS

                    a1,..,aN - algebraics, given in the variables x[N] and t.
[b11,..,b1M],..,[bN1,..,bNM] - lists of algebraics, given in variables x[N]
                               and t.

representing the drift and the diffusion coefficient vectors of the SDE.

stochastic[Taylor1hlf]:=proc(a::list(algebraic),b::list(list(algebraic)))
local u,i,soln;
    for i to nops(a) do
        soln[i] := Y.i[n+1] = Y.i[n]+a[i]*Delta[n]+1/2*L0(a[i],a,b)*Delta[n]^2+
            sum(’op(j,op(i,b))*Delta*W.j[n]+L0(op(j,op(i,b)),a,b)*I[0,j]+
            LJ(a[i],b,j)*I[j,0]’,’j’ = 1 .. nops(op(1,b)))+
            sum(’sum(’LJ(op(j2,op(i,b)),b,j1)*I[j1,j2]’,
            ’j1’ = 1 .. nops(op(1,b)))’,’j2’ = 1 .. nops(op(1,b)))+sum(
            ’sum(’sum(’LJ(LJ(op(p3,op(i,b)),b,p2),b,p1)*I[p1,p2,p3]’,
            ’p1’ = 1 .. nops(op(1,b)))’,’p2’ = 1 .. nops(op(1,b)))’,
            ’p3’ = 1 .. nops(op(1,b)));
        for u to nops(a) do soln[i] := subs(x[u] = Y.u[n],soln[i]) od
    od;
    RETURN(eval(soln))
end:

The call Taylor1hlf([a1,..,aN],[[b11,..,b1M],..,[bN1,..,bNM]]); returns the strong order 1.5 approximation for an N–dimensional SDE (6.1) with M–dimensional noise which has drift coefficients a1, . . ., aN and diffusion coefficient vectors [b11, . . . , b1M], . . ., [bN1, . . . , bNM]. The output consists of the variables Y N[n], DeltaW M[n], Delta[n], I[(j1, j2)], and I[(j1, j2, j3)]. Here Y N[n] denotes the strong order 1.5 stochastic Taylor approximation to x[N] at the n-th step, DeltaW M[n] denotes the change in the M–dimensional Wiener process at the n-th step, Delta[n] denotes thestep size at the n-th step. whileI[(j1, j2)] and I[(j1, j2, j3)] denote the multiple Ito integrals (6.4) and (6.6).

This routine is part of the stochastic package and is usually loaded via the call with(stochastic). It can also be invoked directly via the call stochastic[Taylor1hlf].

EXAMPLE: Consider the 2–dimensional SDE driven by a 2–dimensional Wiener process W_t = (W^1_t , W^2_t ), given by d X^1_t

X^2_t = X^2_t

X^1_t dt + r s dW^1_t + s r dW^2_t , that is with drift components a^1(t, x_1, x_2) = x_2, a^2(t, x_1, x_2) = x_1 and the constant diffusion coefficient vectors

b^1 = b^{1,1} b^{1,2} = r s , b^2 = b^{2,1} b^{2,2} = s r , where r and s are constants.

>Taylor1hlf([x[2],x[1]],[[r,s],[s,r]]);

table([
                                                              2
  1 = (Y1[n + 1] = Y1[n] + Y2[n] Delta[n] + 1/2 Y1[n] Delta[n]
        + r Delta W1[n] + s I[1, 0] + s Delta W2[n] + r I[2, 0]),

                                                              2
  2 = (Y2[n + 1] = Y2[n] + Y1[n] Delta[n] + 1/2 Y2[n] Delta[n]
        + s Delta W1[n] + r I[1, 0] + r Delta W2[n] + s I[2, 0])

      ])

The resulting order 1.5 strong Taylor scheme scheme is

Y^1_{n+1} Y^2_{n+1} = Y^1_n Y^2_n + Y^2_n Y^1_n

Δ_n + 1 2 Y^2_n Y^1_n

(Δ_n)^2 + r s ΔW^1_n

+ s r ΔW^2_n + s r I_{(1,0);n} + r s I_{(2,0);n},

SEE ALSO: stochastic, stochastic[L0], stochastic[LJ], stochastic[Euler], stochastic[Milstein], stochastic[Taylor2] .

6.4 2nd order stochastic Taylor scheme

We now consider the N–dimensional Stratonovich SDE with an M–dimensional Wiener process dX_t = a(t, X_t) dt + M∑ j=1 b^j(t, X_t) ∘ dW^j_t (6.7)

for which the kth component of the order 2.0 strong Taylor scheme is given by

Y^k_{n+1} = Y^k_n + a^k Δ + 1 2 L^0a^k Δ^2 (6.8) + m∑ j=1 ( b^{k,j} ΔW^j + L^0b^{k,j}J_{(0,j)} + L^ja^k J_{(j,0)} )

+ m∑ j_1,j_2=1 ( L^{j1} b^{k,j2} J_{(j1,j2)} + L^0L^{j1} b^{k,j2} J_{(0,j1,j2)}

+L^{j1} L^0b^{k,j2} J_{(j1,0,j2)} + L^{j1} L^{j2} a^k J_{(j1,j2,0)} )

+ m∑ j_1,j_2,j_3=1 L^{j1} L^{j2} b^{k,j3} J_{(j1,j2,j3)}

+ m∑ j_1,j_2,j_3,j_4=1 L^{j1} L^{j2} L^{j3} b^{k,j4} J_{(j1,j2,j3,j4)}.

The J_{(j1,j2)} and J_{(j1,j2,j3)} expressions here denote the corresponding double and triple Stratonovich integrals with respect to the given Wiener process.

The routine stochastic[Taylor2] constructs the stochastic Taylor scheme of strong order 2.0 Taylor scheme for the Stratonovich SDE (6.7). Its CALLING SEQUENCE is

Taylor2([a1,..,aN],[[b11,..,b1M],..,[bN1,..,bNM]]);

with PARAMETERS

                    a1,..,aN - algebraics, given in the variables x[N] and t.
[b11,..,b1M],..,[bN1,..,bNM] - lists of algebraics, given in variables x[N]
                               and t.

represent the drift and the diffusion coefficient vectors of the Stratonovich SDE (6.7).

stochastic[Taylor2]:=proc(a::list(algebraic),b::list(list(algebraic)))
local u,i,soln;
    for i to nops(a) do
        soln[i] := Y.i[n+1] = Y.i[n]+correct(a,b,i)*Delta[n]+
        1/2*SL0(correct(a,b,i),a,b)*Delta[n]^2+
          sum(’op(j,op(i,b))*Delta*W.j[n]+SL0(op(j,op(i,b)),a,b)*J[0,j]+
          LJ(correct(a,b,i),b,j)*J[j,0]’,’j’ = 1 .. nops(op(1,b)))
          +sum(’sum(’LJ(op(j2,op(i,b)),b,j1)*J[j1,j2]+
          SL0(LJ(op(j2,op(i,b)),b,j1),a,b)*J[0,j1,j2]+
          LJ(SL0(op(j2,op(i,b)),a,b),b,j1)*J[j1,0,j2]+
          LJ(LJ(correct(a,b,i),b,j2),b,j1)*J[j1,j2,0]’,
          ’j1’ = 1 .. nops(op(1,b)))’,
          ’j2’ = 1 .. nops(op(1,b)))+sum(
          ’sum(’sum(’LJ(LJ(op(p3,op(i,b)),b,p2),b,p1)*J[p1,p2,p3]’,
          ’p1’ = 1 .. nops(op(1,b)))’,’p2’ = 1 .. nops(op(1,b)))’,
          ’p3’ = 1 .. nops(op(1,b)))+sum(’sum(’sum(
          ’sum(’LJ(LJ(LJ(op(m4,op(i,b)),b,m3),b,m2),b,m1)*J[m1,m2,m3,m4]’,
          ’m1’ = 1 .. nops(op(1,b)))’,’m2’ = 1 .. nops(op(1,b)))
          ’,’m3’ = 1 .. nops(op(1,b)))’,’m4’ = 1 .. nops(op(1,b)));
        for u to nops(a) do soln[i] := subs(x[u] = Y.u[n],soln[i]) od
    od;
    RETURN(eval(soln))
end:

The call Taylor2([a1,..,aN],[[b11,..,b1M],..,[bN1,..,bNM]]); returns the strong order 2.0 stochastic Taylor approximation for the N–dimensional Stratonovich SDE (6.7) with M–dimensional noise which has drift coefficients a1,. . .,aN and diffusion coefficient vectors [b11, . . . , b1M], . . ., [bN1, . . . , bNM].

The output consists of the variables Y N[n], DeltaW M[n], Delta[n], J[(j1, j2)], J[(j1, j2, j3)], and J[(j1, j2, j3, j4)]. here Y N[n] denotes the strong order 2.0 stochastic Taylor approximation to x[N] at the n-th step, DeltaW M[n] denotes the increment in the M–dimensional Wiener process at the n-th step, Delta[n] denotes the step size at the n-th step, while J[(j1, j2)], J[(j1, j2, j3)], and J[(j1, j2, j3, j4)] denote multiple Stratonovich integrals.

This routine is part of the stochastic package and is usually loaded via the call with(stochastic). It can also be invoked directly via the call stochastic[Taylor2]. EXAMPLE: Consider the 2–dimensional Stratonovich SDE driven by a 2–dimensional Wiener process W_t = (W^1_t , W^2_t ), given by d X^1_t X^2_t = X^2_t X^1_t dt + r s ∘ dW^1_t + s r ∘ dW^2_t , that is with drift components a^1(t, x_1, x_2) = x_2, a^2(t, x_1, x_2) = x_1 and the constant diffusion coefficient vectors

b^1 = b^{1,1} b^{1,2} = r s , b^2 = b^{2,1} b^{2,2} = s r , where r and s are constants.

>Taylor2([x[2],x[1]],[[r,s],[s,r]]);

table([
                                                              2
  1 = (Y1[n + 1] = Y1[n] + Y2[n] Delta[n] + 1/2 Y1[n] Delta[n]
      + r Delta W1[n] + s J[1, 0] + s Delta W2[n] + r J[2, 0])
                                                              2
  2 = (Y2[n + 1] = Y2[n] + Y1[n] Delta[n] + 1/2 Y2[n] Delta[n]
      + s Delta W1[n] + r J[1, 0] + r Delta W2[n] + s J[2, 0])

     ])

The resulting order 2.0 strong Stratonovich Taylor scheme scheme is Y^1_{n+1} Y^2_{n+1} = Y^1_n Y^2_n + Y^2_n Y^1_n

Δ_n + 1 2 Y^2_n Y^1_n

(Δ_n)^2 + r s ΔW^1_n

+ s r J_{(1,0);n} + s r ΔW^2_n + r s J_{(2,0);n},

SEE ALSO: stochastic, stochastic[LJ], stochastic[SL0], stochastic[Euler], stochastic[Milstein], stochastic[Taylor1hlf] , stochastic[correct] .

7 Commutative Noise

We consider a the N–dimensional Ito SDE with an M–dimensional Wiener process dX_t = a(t, X_t) dt + M∑ j=1 b^j(t, X_t) dW^j_t . (7.1)

Strongly convergent numerical schemes can be simplified to avoid the need to simulate multiple stochastic integrals when the noise coefficients b^j, j = 1, . . ., M satisfy certain relationships known as commutative noise.

7.1 Commutative noise of 1st kind

The SDE is said to have commutative noise of the first kind when the diffusion coefficients satisfy the condition

L^{j1} b^{k,j2} = L^{j2} b^{k,j1} (7.2)

for all j_1, j_2 = 1, . . ., M, k = 1, . . ., N and (t, x) ∈ ℜ^+ ×ℜ^N .

For instance, additive noise, diagonal noise and linear noise all satisfy this commutativity condition, where diagonal noise means that b^{k,j}(t, x) ≡ 0 and ∂b^{j,j} ∂x^k (t, x) ≡ 0 and linear noise means that b^{k,j}(t, x) = b^{k,j}(t) x^k for all j = 1, . . ., M, k = 1, . . ., N and (t, x) ∈ ℜ^+ ×ℜ^N .

The routine stochastic[comm1] informs the user if the diffusion matrix of an Ito SDE has commutative noise of the first kind. Its CALLING SEQUENCE

comm1([b11,..,b1M],..,[bN1,..,bNM]);

has PARAMETERS

[b11,..,b1M],..,[bN1,..,bNM] - lists of algebraics, given in the variables
                               x[N] and t.

which are the diffusion coefficient vectors of the SDE.

 stochastic[comm1]:=proc()
local LJ1,LJ2,k,j1,j2,flag,p;
    for p to nargs do

   if type(args[p],list) <> true then
        ERROR(‘Expecting input to be an expression sequence of lists‘) fi
    od;
    for k to nargs do
        for j1 to nops(args[1]) do
            for j2 to nops(args[1]) do
                LJ1 := sum(’op(j1,args[l])*diff(op(j2,args[k]),x[l])’,
                ’l’ = 1 .. nargs);
                LJ2 := sum(’op(j2,args[l])*diff(op(j1,args[k]),x[l])’,
                ’l’ = 1 .. nargs);
                if LJ1 <> LJ2 then flag := 1 fi

            od
        od
    od;
    if flag = 1 then
    RETURN(‘Commutative noise of the first kind doesn’t exist for this system‘)
    else RETURN(‘This system exhibits commutative noise of the first kind‘)
    fi;
end:

The call comm1([b11,..,b1M],..,[bN1,..,bNM]); returns a statement indicating whether or not the SDE with this diffusion coefficient matrix has commutative noise of the first kind (7.2). Here [b11, . . . , b1M],. . ., [bN1, . . . , bNM] denotes the M N– dimensional diffusion coefficient vectors, where N is the dimension of the SDE and M is the dimension of the Wiener process.

This routine is part of the stochastic package and is usually loaded via the call with(stochastic). It can also be invoked directly via the call stochastic[comm1]

EXAMPLE: Consider a 2–dimensional Ito SDE with the variable diffusion coefficient vectors

b^1 = b^{1,1} b^{1,2} = 1 0 , b^2 = b^{2,1} b^{2,2} = 0 x_1 .

>comm1([1,0],[0,x[1]]);

Commutative noise of the first kind doesn’t exist for this system

SEE ALSO: stochastic, stochastic[comm2]

7.2 Commutative noise of 2nd kind

Commutative noise of the second kind arises when the noise coefficients satisfy the condition L^{j1} J^{j2} b^{k,j3} = L^{j2} L^{j1} b^{k,j3} (7.3)

for all j_1, j_2, j_3 = 1, . . ., M, k = 1, . . ., N and (t, x) ∈ ℜ^+ ×ℜ^N .

The routine stochastic[comm2] informs the user if the diffusion matrix of an Ito SDE has commutative noise of the second kind. Its CALLING SEQUENCE

comm2([b11,..,b1M],..,[bN1,..,bNM]);

has PARAMETERS

[b11,..,b1M],..,[bN1,..,bNM] - lists of algebraics, given in the variables
                               x[N] and t

which are the difffusion coefficient vectors of the SDE.

stochastic[comm2]:= proc()
local LJ1LJ2,LJ2LJ1,k,p,j1,j2,j3,flag;
    for p to nargs do
        if type(args[p],list) <> true then
        ERROR(‘Expecting input to be an expression sequence of lists‘) fi;
    od;
    for k to nargs do
        for j1 to nops(args[1]) do
            for j2 to nops(args[1]) do
                for j3 to nops(args[1]) do
                    LJ1LJ2 := sum(’op(j1,args[m])*diff(sum(’op(j2,args[l])*
                    diff(op(j3,args[k]),x[l])’,’l’ = 1 .. nargs),x[m])’,
                        ’m’ = 1 .. nargs);
                    LJ2LJ1 := sum(’op(j2,args[m])*diff(sum(’op(j1,args[l])*
                    diff(op(j3,args[k]),x[l])’,’l’ = 1 .. nargs),x[m])’,
                        ’m’ = 1 .. nargs);
                    if LJ1LJ2 <> LJ2LJ1 then flag := 1 fi;
                od;
            od;
        od;
    od;
    if flag = 1 then
    RETURN(‘Commutative noise of the second kind doesn’t exist for this
system‘)
    else RETURN(‘This system exhibits commutative noise of the second kind‘)
    fi;
end:

The call comm2([b11,..,b1M],..,[bN1,..,bNM]); returns a statement indicating whether or not the diffusion matrix of the SDE has commutative noise of the second kind (7.3). Here [b11, . . . , b1M],. . ., [bN1, . . . , bNM] denote the M N–dimensional diffusion coefficient vectors, where N is the dimension of the SDE and M is the dimension of the Wiener process.

This routine is part of the stochastic package and is usually loaded via the call with(stochastic). It can also be invoked directly via the call stochastic[comm2].

EXAMPLE: Consider an 2–dimensional Ito SDE with the variable diffusion coefficient vectors

b^1 = b^{1,1} b^{1,2} = 1 (x_2)^2(x_1)^4

, b^2 = b^{2,1} b^{2,2} = 0 (x_1)^2 .

>comm2([1,(x[2])^2*(x[1])^4],[0,(x[1])^2]);

     Commutative noise of the second kind doesn’t exist for this system

SEE ALSO: stochastic, stochastic[comm1]

8 Weak Numerical Schemes

We consider a the N-dimensional Ito SDE with an M-dimensional Wiener process dX_t = a(t, X_t) dt + M∑ j=1 b^j(t, X_t) dW^j_t . (8.1)

8.1 Weak Euler scheme

The general multi-dimensional case d, m = 1, 2, . . . the kth component of the Euler scheme has the form Y^k_{n+1} = Y^k_n + a^k Δ + m∑ j=1 b^{k,j} ΔW^j, (8.2)

with initial value Y_0 = X_0, where

Δ = τ_{n+1} − τ_n and ΔW^j = W^j_{τn+1} − W^j_{τn} .

For weak convergence only the measure induced by the Ito process X needs to be approximated, so the Gaussian increments ΔW^j in (1.1) can be replaced by simpler random variables ΔŴ^j with similar moment properties giving a simpler scheme by choosing more easily generated noise increments. This leads to the simplified weak Euler scheme

Y^k_{n+1} = Y^k_n + a^k Δ + M∑ j=1 b^{k,j} ΔŴ^j, (8.3)

where the ΔŴ^j for j = 1, 2, . . ., m must be independent A_{τn+1} -measurable random variables with moments satisfying the conditions | | |E ( ΔŴ^j )| | + | | | |E (( ΔŴ^j )3)| | | + | | | |E (( ΔŴ^j )2) − Δ | | | | ≤ K Δ^2

for some constant K; see also (5.12.7). A very simple example of such a ΔŴ^j in (1.2) is a two-point distributed random variable with P ( ΔŴ^j = ± √ Δ ) = 1 2 . The routine stochastic[wkeuler] constructs the stochastic Taylor scheme of weak order 1.0, known as weak Euler schemes. Its CALLING SEQUENCE is

wkeuler([a1,..,aN],[[b11,..,b1M],..,[bN1,..,bNM]]);

with PARAMETERS:

                    a1,..,aN - algebraics, given in the variables x[N] and t.
[b11,..,b1M],..,[bN1,..,bNM] - lists of algebraics, given in variables
                                x[N] and t.

representing the drift and diffusion coefficient vectors of the SDE.

stochastic[wkeuler]:=proc(a::list(algebraic),b::list(list(algebraic)))
local u,i,soln;
    for i to nops(a) do
        soln[i] := Y.i[n+1] = Y.i[n]+L0(x[i],a,b)*Delta[n]+
        sum(’LJ(x[i],b,j)*Delta*Ws.j[n]’,’j’ = 1 .. nops(op(1,b)));
        for u to nops(a) do soln[i] := subs(x[u] = Y.u[n],soln[i]) od
    od;
    RETURN(eval(soln))
end:

The call wkeuler([a1,..,aN],[[b11,..,b1M],..,[bN1,..,bNM]]); returns the simplified weak Euler scheme for an N–dimensional SDE with M–dimensional noise which has drift coefficients a1, . . ., aN and diffusion coefficient vecotrs [b11, . . . , b1M], . . ., [bN1, . . . , bNM]. The output consists of the variables Y N[n], DeltaW sM[n] and Delta[n]. Here Y N[n] denotes the 1st order simplified weak approximation to x[N] at the n-th step, DeltaW sM[n] denotes the increment in the M–dimensional noise process at the n-th step (note here that W sM[n] does not need to denote a standard Wiener processes, but can instead be independent random variables as described above) and Delta[n] denotes the step size at the n-th step.

This routine is part of the stochastic package and is usually loaded via the call with(stochastic). It can also be invoked directly via the call stochastic[wkeuler]. EXAMPLE: Consider the 2–dimensional SDE driven by a 2–dimensional Wiener process W_t = (W^1_t , W^2_t ), given by d X^1_t X^2_t = X^1_t X^2_t dt + r 0 dW^1_t + 0 r dW^2_t , that is with drift components a^1(t, x_1, x_2) = x1, a^2(t, x_1, x_2) = x_2 and the constant diffusion coefficient vectors

b^1 = b^{1,1} b^{1,2} = r 0 , b^2 = b^{2,1} b^{2,2} = 0 r , where r is a constant.

>wkeuler([x[1],x[2]],[[r,0],[0,r]]);

table([
            1 = (Y1[n + 1] = Y1[n] + Y1[n] Delta[n] + r Delta Ws1[n])

            2 = (Y2[n + 1] = Y2[n] + Y2[n] Delta[n] + r Delta Ws2[n])
     ])

The resulting simplified weak Euler scheme is

Y^1_{n+1} Y^2_{n+1} = Y^1_n Y^2_n + Y^2_n Y^1_n

Δ_n + r 0 ΔŴ^1_n + 0 r ΔŴ^2_n .

SEE ALSO: stochastic, stochastic[LO], stochastic[LJ],stochastic[wktay2] .

8.2 Second order weak Taylor scheme

The kth component of the order 2.0 weak Taylor scheme takes the form

Y^k_{n+1} = Y^k_n + a^k Δ + 1 2 L^0a^k Δ^2 (8.4) + m∑ j=1 { b^{k,j}ΔW^j + L^0b^{k,j} I_{(0,j)} + L^ja^k I_{(j,0)} }

+ m∑ j_1,j_2=1 L^{j1} b^{k,j2} I_{(j1,j2)}. Here multiple Ito integrals involving different components of the Wiener process are used. These are generally not easy to generate, so the above scheme is more of theoretical interest than of practical use. However, for weak convergence we can substitute simpler random variables for the multiple Ito integrals to obtain the following simplified order 2.0 weak Taylor scheme with kth component

Y^k_{n+1} = Y^k_n + a^k Δ + 1 2 L^0a^k Δ^2 (8.5) + m∑ j=1 { b^{k,j} + 1 2 Δ ( L^0b^{k,j} + L^ja^k )} ΔŴ^j

+ 1 2 m∑ j_1,j_2=1 L^{j1} b^{k,j2} ( ΔŴ^{j1} ΔŴ^{j2} + V_{j1,j2} ) .

Here the ΔŴ^j for j = 1, 2, . . ., m are independent random variables satisfying the moment conditions| | |E ( ΔŴ )| | + | | |E (( ΔŴ )3)| | + | | |E (( ΔŴ )5)| | (8.6) + | | | |E (( ΔŴ )2) − Δ | | | | + | | | |E (( ΔŴ )4) − 3Δ^2 | | | | ≤ K Δ^3

for some constant K and the V_{j1,j2} are independent two-point distributed random variables with P (V_{j1,j2} = ±Δ) = 1 2 (8.7)

for j_2 = 1, . . ., j_1 − 1, V_{j1,j1} = −Δ (8.8) and V_{j1,j2} = −V_{j2,j1} (8.9)

for j_2 = j_1 + 1, . . ., m and j_1 = 1, . . ., m. Here we can use a three-point distributed random variable ΔŴ with P ( ΔŴ = ± √ 3Δ ) = 1 6 , P ( ΔŴ = 0 ) = 2 3 . (8.10)

The routine stochastic[wktay2] constructs simplified stochastic Taylor schemes of weak order 2.0. Its CALLING SEQUENCE is

wktay2([a1,..,aN],[[b11,..,b1M],..,[bN1,..,bNM]]);

with PARAMETERS:

                    a1,..,aN - algebraics, given in the variables x[N] and t.
[b11,..,b1M],..,[bN1,..,bNM] - lists of algebraics, given in variables
                                x[N] and t.

representing the drift and diffusion coefficient vectors of the SDE.

stochastic[wktay2]:=
proc(a::list(algebraic),b::list(list(algebraic)))
local u,i,soln;
    for i to nops(a) do
        soln[i] := Y.i[n+1] = Y.i[n]+a[i]*Delta[n]+1/2*L0(a[i],a,b)*Delta[n]^2+
            sum(’(op(j,op(i,b))+1/2*Delta[n]*(L0(op(j,op(i,b)),a,b)+
            LJ(a[i],b,j)))*Delta*Ws.j[n]’,’j’ = 1 .. nops(op(1,b)))+1/2*
            sum(’sum(’LJ(op(j2,op(i,b)),b,j1)*(Delta^2*Ws.j1[n]*Ws.j2[n]+
            V[j1,j2])’,’j1’ = 1 .. nops(op(1,b)))’,
            ’j2’ = 1 .. nops(op(1,b)));
        for u to nops(a) do soln[i] := subs(x[u] = Y.u[n],soln[i]) od
    od;
    RETURN(eval(soln))
end:

The call wktay2([a1,..,aN],[[b11,..,b1M],..,[bN1,..,bNM]]); returns the simplified weak order 2.0 stochastic Taylor scheme for an N–dimensional SDE with M– dimensional noise which has drift coefficients a1, . . .,aN and diffusion coefficient vec- tors [b11, . . . , b1M],. . ., [bN1, . . . , bNM]. The output consists of the variables Y N[n], DeltaW sM[n], V [(j1, j2)], and Delta[n]. Here Y N[n] denotes the 2nd order simplified weak approximation to x[N] at the n-th step, DeltaW sM[n] denotes the change in the M–dimensional noise process at the n-th step (note here that W sM[n] does not denote standard Wiener processes, but instead are independent random variables described above), V [(j1, j2)] denotes the independent two-point random variables described above, and Delta[n] denotes the step size at the n-th step.

This routine is part of the stochastic package and is usually loaded via the call with(stochastic). It can also be invoked via the call stochastic[wktay2].

EXAMPLE: Consider the 2–dimensional SDE driven by a 2–dimensional Wiener process W_t = (W^1_t , W^2_t ), given by d X^1_t

X^2_t = X^1_t

X^2_t dt + r 0 dW^1_t + 0 r dW^2_t , that is with drift components a^1(t, x_1, x_2) = x_1, a^2(t, x_1, x_2) = x_2 and the constant diffusion coefficient vectors

b^1 = b^{1,1} b^{1,2} = r 0 , b^2 = b^{2,1} b^{2,2} = 0 r , where r is a constant.

>wktay2([x[1],x[2]],[[r,0],[0,r]]);

table([
                                                                      2
          1 = (Y1[n + 1] = Y1[n] + Y1[n] Delta[n] + 1/2 Y1[n] Delta[n]
                                 + (r + 1/2 Delta[n] r) Delta Ws1[n])

                                                                     2
         2 = (Y2[n + 1] = Y2[n] + Y2[n] Delta[n] + 1/2 Y2[n] Delta[n]
                                + (r + 1/2 Delta[n] r) Delta Ws2[n])
     ])

The resulting weak order 2.0 Taylor scheme is Y^1_{n+1} Y^2_{n+1} = Y^1_n Y^2_n + Y^1_n Y^2_n

Δ_n + 1 2 Y^1_n Y^2_n

(Δ_n)^2

+ r +^1_2 rΔ_n 0 ΔŴ^1_n + 0 r +^1_2 rΔ_n

ΔŴ^2_n .

SEE ALSO: stochastic, stochastic[L0], stochastic[LJ], stochastic[wkeuler] .

8.3 Order 3 weak Taylor scheme

The kth component of the order 3.0 weak Taylor scheme takes the form

Y^k_{n+1} = Y^k_n + a^k Δ + m∑ j=1 b^{k,j}ΔW^j + m∑ j=0 L^ja^k I_{(j,0)} (8.11)

+ m∑ j_1=0 m∑ j_2=1

L^{j1} b^{k,j2} I_{(j1,j2)} + m∑ j_1,j_2=0

L^{j1} L^{j2} a^k I_{(j1,j2,0)}

+ m∑ j_1,j_2=0 m∑ j_3=1 L^{j1} L^{j2} b^{k,j3} I_{(j1,j2,j3)}. Various simplifications are possible in special cases that avoid the need to generate the multiple stochastic integrals. See Chapter 14.3 of Kloeden and Platen [7] The routine stochastic[wktay3] constructs stochastic Taylor schemes of weak order 3.0. Its CALLING SEQUENCE is wktay3([a1,..,aN],[[b11,..,b1M],..,[bN1,..,bNM]]);

with PARAMETERS:

                    a1,..,aN - algebraics, given in the variables x[N] and t.
[b11,..,b1M],..,[bN1,..,bNM] - lists of algebraics, given in variables
                                x[N] and t.

representing the drift and diffusion coefficient vectors of the SDE.

stochastic[wktay3]:=proc(a::list(algebraic),b::list(list(algebraic)))
local u,i,soln;
    for i to nops(a) do
        soln[i] := Y.i[n+1] = Y.i[n]+a[i]*Delta[n]+
        sum(’op(j,op(i,b))*Delta*W.j[n]’,’j’ = 1 .. nops(op(1,b)))+
         sum(’MLJ(a[i],a,b,j0)*I[j0,0]’,’j0’ = 0 .. nops(op(1,b)))+
         sum(’sum(’MLJ(op(j2,op(i,b)),a,b,j1)*I[j1,j2]’,
         ’j2’ = 1 .. nops(op(1,b)))’,’j1’ = 0 .. nops(op(1,b)))+
         sum(’sum(’MLJ(MLJ(a[i],a,b,k2),a,b,k1)*I[k1,k2,0]’,
         ’k1’ = 0 .. nops(op(1,b)))’,’k2’ = 0 .. nops(op(1,b)))+sum(
         ’sum(’sum(’MLJ(MLJ(op(m3,op(i,b)),a,b,m2),a,b,m1)*I[m1,m2,m3]’,
         ’m3’ = 1 .. nops(op(1,b)))’,’m2’ = 0 .. nops(op(1,b)))’,
         ’m1’ = 0 .. nops(op(1,b)));
        for u to nops(a) do soln[i] := subs(x[u] = Y.u[n],soln[i]) od
    od;
    RETURN(eval(soln))
end:

SYNOPSIS: The call wktay3([a1,..,aN],[[b11,..,b1M],..,[bN1,..,bNM]]); returns the weak order 3.0 stochastic Taylor scheme for an N–dimensional SDE with M– dimensional noise which has drift coefficients a1, . . ., aN and diffusion coefficient vectors [b11, . . . , b1M], . . ., [bN1, . . . , bNM]. The output consists of the variables Y N[n], DeltaW M[n], I[(j1, j2)], I[(j1, j2, j3)] and Delta[n]. Here Y N[n] denotes the 3rd order weak approximation to x[N] at the n-th step, DeltaW M[n] denotes the increment in the M–dimensional Wiener process at the n-th step, I[(j1, j2)] and I[(j1, j2, j3)] denote multiple Ito integrals described above, and Delta[n] denotes the step size at the n-th step.

This routine is part of the stochastic package and is usually loaded via the call with(stochastic). It can also be invoked directly via the call stochastic[wktay3].

EXAMPLE: Consider the 2–dimensional SDE driven by a 2–dimensional Wiener process W_t = (W^1_t , W^2_t ), given by d X^1_t

X^2_t = X^2_t

X^1_t dt + r s dW^1_t + s r dW^2_t , that is with drift components a^1(t, x_1, x_2) = x_2, a^2(t, x_1, x_2) = x_1 and the constant diffusion coefficient vectors

b^1 = b^{1,1} b^{1,2} = r s , b^2 = b^{2,1} b^{2,2} = s r , where r and s are constants.

>wktay3([x[2],x[1]],[[r,s],[s,r]]);

table([
  1 = (Y1[n + 1] = Y1[n] + Y2[n] Delta[n] + r Delta W1[n] + s Delta W2[n]
                     + Y1[n] I[0, 0] + s I[1, 0] + r I[2, 0] + Y2[n] I[0, 0, 0]
                     + r I[1, 0, 0] + s I[2, 0, 0]),

  2 = (Y2[n + 1] = Y2[n] + Y1[n] Delta[n] + s Delta W1[n] + r Delta W2[n]
                     + Y2[n] I[0, 0] + r I[1, 0] + s I[2, 0] + Y1[n] I[0, 0, 0]
                     + s I[1, 0, 0] + r I[2, 0, 0])‘,
     ])

The resulting order 3.0 weak Taylor scheme scheme is

Y^1_{n+1} Y^2_{n+1} = Y^1_n Y^2_n + Y^2_n Y^1_n

Δ_n + r s ΔW^1_n + s r ΔW^2_n

+ Y^1_n Y^2_n

I_{(0,0);n} + s r I_{(1,0);n} + r s I_{(2,0);n}

+ Y^1_n Y^2_n I_{(0,0,0);n}, + s r I_{(1,0,0);n} + r s I_{(2,0,0);n}

SEE ALSO: stochastic, stochastic[L0], stochastic[MLJ],stochastic[wkeuler], stochastic[wktay2].

9 APPENDIX

9.1 Subprocedures for momenteqn

The following procedures are subprocedures for the calculation or transformtion parts of (5.3). For example, the procedure position determines the position in the new vector and the procedure ap transforms the product AP in a vector equation.

position:=proc(N,i,j)
  global stelle;
  stelle:=sum(’N-k+1’,’k’=1..i-1)+j-i+1;
end:

ap:=proc(A)

  local i,j,k,Atmp,N,counter;
  global B1;
  if type(A,array) then Atmp:=convert(A,listlist);else Atmp:=A; fi;
  N:=nops(Atmp);
  B1:=array(1..N*(N+1)/2,1..N*(N+1)/2);
  for i from 1 to N*(N+1)/2 do
   for j from 1 to N*(N+1)/2 do
     B1[i,j]:=0;
   od;
  od;
  counter:=0:
  for i from 1 to N do
   for j from i to N do
     counter:=counter+1;
     for k from 1 to N do
       if (j<=k) then B1[counter,position(N,j,k)]:=A[i,k];
                else B1[counter,position(N,k,j)]:=A[i,k];
       fi;
     od;
   od;
  od;
  RETURN(evalm(B1));
end:

pa:=proc(A)
  local i,j,k,Atmp,N,counter;
  global B2;
  if type(A,array) then Atmp:=convert(A,listlist);else Atmp:=A; fi;
  N:=nops(Atmp);
  B2:=array(1..N*(N+1)/2,1..N*(N+1)/2);
  for i from 1 to N*(N+1)/2 do
   for j from 1 to N*(N+1)/2 do
     B2[i,j]:=0;
   od;
  od;
  counter:=0:
  for i from 1 to N do
   for j from i to N do
     counter:=counter+1;
     for k from 1 to N do
       if (i<=k) then B2[counter,position(N,i,k)]:=A[j,k];
                else B2[counter,position(N,k,i)]:=A[j,k];
       fi;
     od;
   od;
  od;
  RETURN(evalm(B2));
end:

bpb:=proc(B)
  local i,j,k,l,Btmp,N,counter;
  global B3;
  if type(B,array) then Btmp:=convert(B,listlist);else Btmp:=B; fi;
  N:=nops(Btmp);
  B3:=array(1..N*(N+1)/2,1..N*(N+1)/2);

  for i from 1 to N*(N+1)/2 do
   for j from 1 to N*(N+1)/2 do
     B3[i,j]:=0;
   od;
  od;
  counter:=0:
  for i from 1 to N do
   for j from i to N do
     counter:=counter+1;
     for l from 1 to N do
       for k from 1 to N do
         if (k<=l) then
B3[counter,position(N,k,l)]:=B3[counter,position(N,k,l)]
                       +B[i,k]*B[j,l];
                  else
B3[counter,position(N,l,k)]:=B3[counter,position(N,l,k)]
                       +B[i,k]*B[j,l];
         fi;
       od;
     od;
   od;
  od;
  RETURN(evalm(B3));
end:

9.2 The inverse procedure pvector2pmatrix

The procedure pvector2pmatrix is the inverse of the procedure matrix2pvector. It transforms a vector to an symmetry matrix and has following code:

pvector2pmatrix:=proc(pvector)
  local i,j,k,ptmp,N;
  global p;
  if type(pvector,array) then ptmp:=convert(pvector,list);else
  ptmp:=pvector; fi;
  N:=-1/2+sqrt(1/4+2*nops(ptmp));
  p:=array(1..N,1..N);
  k:=0;
  for i from 1 to N do
    if (i>1) then k:=k+(N-i+2); fi;
    for j from i to N do
      p[i,j]:=ptmp[k+j-i+1];
      if (i<>j) then p[j,i]:=p[i,j]; fi;
    od;
  od;
  RETURN(eval(p));
end:

EXAMPLE : Consider the 15-dimensional vector

p̃ = [1, 2, 3, 4, 5, 6, 7, 8, 9, 10, 11, 12, 13, 14, 15].

The result of pvector2pmatrix( p̃); gives the 5 × 5 symmetric matrix P .

1 2 3 4 5 2 6 7 8 9 3 7 10 11 12 4 8 11 13 14 5 9 12 14 15 .

References

[1] R.E. Crandall, Topics in Advanced Scientific Computation, Springer–Verlag, Heidelberg (1996)

[2] S.O. Cyganowski, Solving Stochastic Differential Equations with Maple, MapleTech Newsletter 3(2) (1996), 38–.

[3] S.O. Cyganowski, A MAPLE Package for Stochastic Differential Equations, in “Computational Techniques and Applications: CTAC95” (Editors A. Easton, & R. May), World Scientific Publishers, Singapore (1996)

[4] S. Cyganowski and P.E. Kloeden, Stochastic stability examined through MAPLE, in Proc. 15th IMACS World Congress, Volume 1: Computational Mathematics (Editor: A. Sydow), Wissenschaft & Technik Verlag, Berlin, 1997, 4372–4377.

[5] W. Gander and J. Hrebicek, Solving Problems in Scientific Computing using Maple and Matlab, Second Eddition, Springer–Verlag, Heidelberg (1995)

[6] W.S. Kendall, Computer algebra and stochastic calculus, Notices Amer. Math. Soc.

37 (1990), 1254– .

[7] P.E. Kloeden and E. Platen, Numerical Solution of Stochastic Differential Equations Springer-Verlag, Heidelberg (1992)

[8] P.E. Kloeden, E. Platen and H. Schurz, Numerical Solution of Stochastic Differential Equations through Computer Experiments, Springer-Verlag, Heidelberg (1993).

[9] P.E. Kloeden and W.D. Scott, Construction of Stochastic Numerical Schemes through Maple, MapleTech Newsletter 10 (1993), 60–65.

[10] G.G. Milshtein and M.V. Tret’yakov, Numerical Solution of Differential Equations with Coloured Noise, J. Stat. Physics, 77 (1994) 691–715.

[11] E. Valkeila, Computer algebra and stochastic analysis, some possibilities, CWI Quarterly 4 (1991), 229–